



DOMINA LAS FUNCIONES DE BÚSQUEDA EN EXCEL

Tabla de contenido

Tabla de contenido	2
¿Por qué deberías leer este libro?	5
Ejemplos	5
¿Cómo está organizado el ebook?	5
BUSCARX.....	6
Sintaxis de BUSCARX.....	6
Ventajas de BUSCARX con respecto a BUSCARV.....	7
Ejemplos BÁSICOS.....	8
BUSCARV.....	12
Sintaxis de BUSCARV	12
Ejemplos básicos.....	13
BUSCARV y los errores #N/D	14
BUSCARH.....	16
Sintaxis de BUSCARH	16
Algunas cosas que debes conocer sobre BUSCARH	17
Ejemplos básicos.....	17
INDICE+COINCIDIR.....	19
La función INDICE	19
La función COINCIDIR	20
Búsquedas combinando INDICE y COINCIDIR	20
Funciones de búsqueda y tablas	22
Cómo crear una tabla de Excel	22
¿Qué ocurre al crear una tabla?.....	23
Sobre las referencias estructuradas	24
¿Por qué usar referencias estructuradas en lugar de referencias de rango?..	24
Otras utilidades de las tablas.....	25
Uso de comodines	26
Validación de datos con funciones de búsqueda.....	28
Pedro, el empleado con el trabajo más aburrido del mundo	28
¿Cómo están estructurados los datos de Pedro?	28

Las funciones ENCONTRAR y HALLAR.....	33
La función ENCONTRAR	33
La función HALLAR	34
Ejemplos con funciones de búsqueda	35
BUSCARX con comodines	35
BUSCARX con resultado matricial en diferente orden	37
BUSCARX en filas o columnas	38
BUSCARX sensible a mayúsculas y minúsculas	39
Meses con mayor y menor venta	40
buscarv EN DOS TABLAS SEGÚN un criterio	41
Calcular importe del envío con BUSCARV	42
BUSCARV hacia la izquierda.....	42
BUSCARV en otra hoja u otro libro	44
BUSCARV en tablas de doble entrada	45
Fusionar varias tablas	46
BUSCARH para encontrar el valor de la primera celda no vacía	47
Mostrar resultados de búsqueda por categoría de forma vertical	48
Buscar valores con varias condiciones	51
Varias técnicas de búsqueda de datos en tablas de doble entrada.....	57
Cómo encontrar la 2ª, 3ª o nª coincidencia	60
Algunos trucos mostrados en el ebook	62
Utiliza rangos con nombres	62
Utiliza referencias de celdas absolutas siempre que sea posible	63
Utiliza tablas	63
Gestiona los errores	63
Utiliza BUSCARV solo cuando lo necesites	64
¿Quieres aprender más sobre Excel?	65
Descarga de archivos	65

Introducción

¡Hola! Mi nombre es Sergio Propergol y llevo al frente de Ayuda Excel desde 2006. Si no sabes cómo he llegado hasta aquí, ayudando a centenares de personas y empresas a optimizar su trabajo con Excel, puedes seguir leyendo. Si ya me conoces y quieres ir directo al grano, puedes pasar al primer capítulo.



Aunque llevo trabajando con Excel desde el año 2000, no fue hasta 2006 cuando creé Ayuda Excel. Por aquel entonces yo también tenía muchas dudas con la hoja de cálculo y quería mejorar mis habilidades para ser mejor en mi trabajo. Junto con otros usuarios que se unieron rápidamente al foro de la web, fuimos resolviendo nuestras propias dudas, ayudándonos unos a otros, y ayudando a cientos de usuarios de usuarios de todo el mundo.

Cuanto más aprendía, más potencial le veía a Excel como herramienta, así que, comencé a formarme de forma avanzada hasta que en 2008 lancé en la web mi servicio de consultoría profesional para ofrecer a particulares y empresas varios tipos de soluciones para sus hojas de cálculo. Desde entonces he trabajado para empresas como I.C.O.N., Dufry, Anaya, Fundación Biodiversidad y para grandes profesionales de los negocios online como Antonio G (Inteligencia Viajera, Escuela Nómada Digital), Ángel Alegre (Vivir al Máximo) o Javier Barros.

A día de hoy, el foro de Ayuda Excel es el foro de Excel más grande de habla hispana, con más de 25.600 usuarios de todo el mundo y la web fue premiada con el MVP de Microsoft 5 años consecutivos.

Desde que comencé a trabajar con Excel, me he sentido atraído por las búsquedas de datos. Seguramente es porque mi trabajo estaba basado en el formateo y transformación de textos (en esa época no existía Power Query).

Más de 20 años después sigo haciendo búsquedas (no con tanta intensidad como antes) pero siempre llega un momento en el que debo extraer cierto dato de una tabla, una hoja o una base de datos.

Estoy seguro de que no soy el único que realiza búsquedas regularmente, ya que Excel es una gran herramienta para gestionar y analizar datos. Si estás leyendo este texto desde tu ordenador, tablet o teléfono, es porque necesitas encontrar una forma más fácil y automática de buscar un producto en una tabla, conocer el número de veces que se repite un carácter en el contenido de una celda, encontrar la categoría de un empleado para aplicarle un bonus en su nómina, hallar una lista de valores que comienzan con una determinada secuencia de caracteres.... En fin, las posibilidades son más que infinitas.

¿Qué encontrarás en este libro?

El objetivo de este libro es ayudarte a dominar las funciones de búsqueda en Excel, mucho más allá de BUSCARV o BUSCARH. Conocerás todas las opciones de búsqueda que puedes utilizar, aprenderás a usarlas, a diferenciarlas y a elegir la opción que mejor te convenga según tus necesidades.

Además, he incluido trucos que te harán la vida mucho más fácil al trabajar con búsquedas y muchos ejemplos que harán que veas el funcionamiento de las búsquedas de una forma mucho más sencilla y visual.

Ejemplos

He incluido varios archivos con ejemplos junto con el e-book. **Puedes encontrar el enlace de descarga en la última página.**

¿Cómo está organizado el e-book?

El libro se divide en dos partes. En la primera describiré las funciones de búsqueda de forma individual e incluyo unos ejemplos sencillos para practicar con ellas.

La segunda parte se enfoca en búsquedas más avanzadas, donde te demostraré que combinando dos o más funciones es posible realizar cualquier búsqueda imaginable.

Prepárate porque ¡comenzamos!

BUSCARX

BUSCARX es la función de Excel más reciente relacionada con búsquedas. Se trata de una versión muy mejorada de las funciones BUSCARV y BUSCARH.

La misión de BUSCARX es prácticamente la misma que BUSCARV: permite buscar un valor en un conjunto de datos (vertical u horizontal) y devuelve el valor correspondiente en alguna fila o columna.

Por ejemplo, si tienes un listado con las notas de los alumnos obtenidas en un examen, podrías usar BUSCARX para encontrar la nota de uno o varios alumnos utilizando únicamente su nombre.


A simple vista no tiene ninguna diferencia con otras funciones de búsqueda, pero a lo largo de los ejemplos del manual irás viendo toda su potencia y flexibilidad.

En el momento de escribir este manual (julio de 2020), BUSCARX solo está disponible para los usuarios de Office 365. Por lo tanto, si estás utilizando Excel 2019, 2016, 2013 o anteriores, no podrás utilizarla. Es tanta la importancia que tiene esta función que quizá (no estoy seguro) llegue a publicarse para versiones anteriores en forma de complemento (como ya sucedió con la publicación de Power Query).

De momento solo se puede utilizar si tienes instalado Office 365 o si utilizas Excel desde la web.

Sintaxis de BUSCARX

La sintaxis de BUSCARX es la siguiente:



Los argumentos entre corchetes son opcionales:

- **Valor_buscado:** El valor a buscar.
- **Matriz_buscada:** La tabla o rango de celdas donde se buscará.
- **Matriz_devuelta:** La matriz de celdas que se devolverá.

- **Si_no_se_encuentra:** Valor a devolver en caso de que el valor no se encuentre.
- **Modo_de_coincidencia:** Especifica qué tipo de coincidencia se debe buscar:
 - **0:** Coincidencia exacta. Si no se encuentra ninguna coincidencia se devuelve el error #N/D (o #N/A, según la versión de Excel). Es el valor predeterminado.
 - **-1:** Coincidencia exacta o el siguiente elemento menor. Si no se encuentra el valor devuelve el siguiente elemento más pequeño.
 - **1:** coincidencia exacta o el siguiente elemento mayor. Si no se encuentra el valor devuelve el siguiente elemento más grande.
 - **2:** Coincidencia de caracteres comodín. Se utiliza para hacer búsquedas parciales con caracteres comodín (*, ? o ~).
- **Modo_de_búsqueda:** Especifica el modo de búsqueda que se usará.
 - **1:** Búsqueda del primero al último. Realiza la búsqueda comenzando por el primer elemento (de arriba a abajo). Es el valor predeterminado.
 - **-1:** Búsqueda del último al primero. Realiza la búsqueda comenzando por el último elemento (de abajo a arriba).
 - **2:** Búsqueda binaria (orden ascendente). Es necesario que los datos estén ordenados de forma ascendente.
 - **-2:** Búsqueda binaria (orden descendente). Es necesario que los datos estén ordenados de forma descendente.

Ventajas de BUSCARX con respecto a BUSCARV

Son numerosas las ventajas que tiene BUSCARX con respecto a BUSCARV. Por eso pienso que es una versión muy mejorada de esta última. Sus principales ventajas son las siguientes:

- Con BUSCARX es posible hacer búsquedas hacia la izquierda.
- BUSCARX seguirá funcionando si se insertan columnas o filas en la matriz de búsqueda.
- BUSCARX puede encontrar la última coincidencia de un valor.
- BUSCARX funciona con coincidencia exacta de forma predeterminada. Esto impide que la búsqueda devuelva accidentalmente datos erróneos.
- BUSCARX puede devolver una celda o una matriz.
- BUSCARX permite especificar un valor a devolver en caso de que el valor buscado no se encuentre. Ya no hace falta utilizar SI.ERROR.

Ejemplos BÁSICOS

Ejemplo 1: búsqueda exacta

Como he comentado, BUSCARX realiza búsquedas exactas de forma predeterminadas. ¿Qué significa “exacta”? Simplemente que, si no se encuentra el valor, la función devuelve el error #N/D.

Esta es la forma más fácil de usar BUSCARX:

	A	B	C	D	E	F	G	H	I	J
1	BUSCARX: búsqueda exacta									
2										
3										
4										
5		Película	Año	Recaudación (millones \$)			Película	Recaudación		
6		Psicosis	1960	60			Los pájaros	=BUSCARX(G6;B6:B11;D6:D11)		
7		Vértigo	1958	7						
8		La ventana Indiscreta	1954	37						
9		Los pájaros	1963	11,4						
10		Con la muerte en los talones	1959	9,8						
11		El hombre que sabía demasiado	1956	11,3						
12										
13										
14										
15										

=BUSCARX(G6;B6:B11;D6:D11)

En el ejemplo se busca el valor de la celda G6 en la matriz B6:B11 y devuelve el valor que se encuentra en la misma posición de la matriz D6:D11.

Ejemplo 2: Búsqueda aproximada

La búsqueda aproximada es útil cuando el dato a buscar no es un dato exacto. En el siguiente ejemplo se utiliza BUSCARX para calcular un descuento en función de la cantidad pedida. Es impensable tener una tabla donde se almacenen todos los posibles valores a buscar. Por eso se usa la búsqueda aproximada.

Para habilitar este tipo de búsqueda debes introducir el argumento **Modo_de_coincidencia**. Observa el ejemplo:

G6 =BUSCARX(F6;B6:B10;C6:C10;;-1)

BUSCARX: búsqueda aproximada						
1						
2						
3						
4						
5		Cantidad	Descuento		Cantidad	Descuento
6		1	0%		32	10%
7		10	5%			
8		25	10%			
9		50	15%			
10		100	20%			
11						
12						
13						

=BUSCARX(F6;B6:B10;C6:C10;;-1)

La fórmula busca el número 32 en la **matriz_de_búsqueda**. Como no existe devuelve el porcentaje de la cantidad inmediatamente inferior (-1). Sería posible obtener la cantidad superior cambiando el argumento **Modo_de_coincidencia** a 1.

Ejemplo 3: resultado matricial

En muchas ocasiones es necesario obtener más de un resultado a la vez para el mismo **Valor_buscado**. BUSCARX elimina la necesidad de introducir varias fórmulas para obtener el registro completo. Observa el siguiente ejemplo. En él se buscan las calificaciones obtenidas por un determinado alumno en todas las materias estudiadas:

C5 =BUSCARX(B5;B9:B30;C9:G30)

BUSCARX: resultado matricial (varios valores)							
1							
2							
3							
4		Alumno	Matemáticas	Física	Historia	Ciencias	Excel
5		Salomón Campoo	6,55	4,56	1,82	6,13	6,32
6							
8		Alumno	Matemáticas	Física	Historia	Ciencias	Excel
9		Adalberto Olid	5,15	4,74	3,36	5,59	8,45
10		Noé Banyuls	9,38	8,88	6,30	7,40	9,61
11		Salomón Campoo	6,55	4,56	1,82	6,13	6,32
12		Xaria Goyeneche	1,70	7,13	4,37	9,68	4,57
13		Olimpia Llamas	4,64	4,75	3,69	8,61	8,10
14		Florida Sacristán	2,84	3,68	6,30	4,79	8,18

=BUSCARX(B5;B9:B30;C9:G30)

Al introducir una única fórmula en la celda C5, el resto de resultados se rellenan automáticamente.

Ejemplo 4: tablas de doble entrada

BUSCARX también se puede utilizar para hacer búsquedas bidireccionales anidando una función en otra. El siguiente ejemplo muestra cómo buscar un resultado en una tabla de doble entrada:

	A	B	C	D	E			
1	BUSCARX: tablas de doble entrada							
2								
3								
4								
5		A	B	C	D	E	Fila	Araceli
6	Atanasio	18	22	16	16	19	Columna	D
7	Blas	33	20	34	29	22	Resultado	30
8	Cristian	19	30	29	33	36		
9	Araceli	28	24	31	30	17		
10	Crisanto	32	15	27	16	31		
11	Fermin	20	30	36	34	31		
12	Xabier	18	25	16	29	29		
13	Urania	25	36	35	19	23		
14	Reina	27	21	22	33	25		
15	Virgilio	23	27	34	19	25		
16	Belinda	23	30	36	33	21		
17	Katia	24	15	21	28	32		
18								

=BUSCARX(J5;B6:B17;BUSCARX(J6;C5:G5;C6:G17))

La fórmula busca el nombre “Araceli”. La **matriz_devuelta** viene determinada por otra función BUSCARX, que hace una búsqueda vertical de “D”, dando como resultado el valor de la intersección entre la fila y la columna.

Ejemplo 5: No se encuentra

Si BUSCARX no encuentra el **valor_buscado** se comporta igual que lo hacen otras funciones de búsqueda: devuelve un error #N/D (#N/A en muchas versiones de Excel).

Para mostrar un mensaje personalizado en lugar del error, introduce un valor en el argumento **si_no_se_encuentra**. Fíjate en el ejemplo:

	A	B	C	D	E	F	G	H
1	BUSCARX: si no se encuentra							
2								
3								
4								
5		Película	Año	Recaudación (millones \$)			Película	Recaudación
6		Psicosis	1960	60			Robocop	No se encuentra
7		Vértigo	1958	7				
8		La ventana Indiscreta	1954	37				
9		Los pájaros	1963	11,4				
10		Con la muerte en los talones	1959	9,8				
11		El hombre que sabía demasiado	1956	11,3				
12								
13								
14								
15								

Si omites el argumento **si_no_se_encuentra**, se mostrará el error #N/D.

La principal ventaja es que se evita el utilizar otras funciones como SI.ERROR para gestionar los posibles errores.

Más adelante en el manual, encontrarás más ejemplos de uso de BUSCARX. Ahora te presentaré otras funciones igual de útiles con las que puedes realizar cualquier tarea de búsqueda que te propongas.

BUSCARV

BUSCARV es mi función favorita. Quizá se deba a que fue una de las primeras que aprendí a utilizar cuando comencé mi vida laboral. Nada más comenzar a trabajar con ella me di cuenta del gran valor que tendría en mi trabajo.

Y a ti te pasará algo similar. Una vez que aprendas a utilizar BUSCARV intentarás encontrar la forma de usarla todo el tiempo.

Existen dos formas de usar BUSCARV. La mayoría de las personas solo conocen una de ellas (coincidencia exacta), pero también es posible utilizarla para hallar coincidencias aproximadas. En este manual te enseñaré a utilizar ambas para que puedas sacarle más partido.

El objetivo de BUSCARV es buscar un determinado dato en la primera columna de una tabla o matriz y, una vez localizada la fila en la que se encuentra, devuelve el valor que se encuentra a la derecha un número determinado de columnas.

Sintaxis de BUSCARV

BUSCARV(valor_buscado;matriz_buscar_en;indicador_columnas;[ordenado])

- **Valor_buscado:** El valor que se desea buscar.
- **Matriz_buscar_en:** La matriz o rango de celdas donde se desea buscar.
- **Indicador_columnas:** La posición de la columna que se quiere devolver.
- **Ordenado:** Opcional. Se indica VERDADERO para búsquedas exactas, FALSO para aproximadas.

Esta sintaxis se ve mucho mejor en la práctica. Observa la siguiente imagen:

3. y devuelve el contenido de la misma fila en la tercera columna

1. Encuentra una coincidencia exacta para este valor

2. En este rango de celdas

Referencia	Descripción	Precio
A10	Manzanas	1,12
B15	Ciruelas	2,21
A24	Naranjas	2,44
C11	Melocotones	1,97
S05	Pineles	2,54
A14	Kiwis	1,68
A1	Cerezas	3,60
B22	Peras	2,69
C42	Mango	2,77

Criterio: S05
Fórmula: =BUSCARV(H5;B6:D14;3;FALSO)
Resultado: 2,54

Utilizando la siguiente fórmula es posible extraer fácilmente el precio de la tabla.

=BUSCARV(H5;B6:D14;3;FALSO)

La tabla de búsqueda aparece con fondo rojo. El **valor_buscado** debe aparecer en la primera columna de la **matriz_buscar_en**. Observa que el **valor_buscado** se encuentra en la celda B10 (en la primera columna de la matriz).

El siguiente paso es decirle a la función qué columna debe devolver (a partir de la primera). Observa que el número 3 de la fórmula no se corresponde con la tercera columna de la hoja, sino con la tercera columna de la **matriz_buscar_en**.


El último argumento, **ordenado**, se ha establecido en FALSO para indicarle a Excel que la búsqueda es exacta, es decir, que si no encuentra el valor, devuelva el error #N/D.

Ejemplos básicos

Ejemplo 1: coincidencia exacta

En la mayoría de las ocasiones querrás utilizar BUSCARV en el modo de coincidencia exacta. Esto tiene sentido cuando estás buscando un valor exacto y buscas el resultado que le corresponde justo a ese valor.

Observa el siguiente ejemplo: Es absurdo buscar el año de producción de una película que no existe. Imagina buscar la película y que el resultado sea el año de la película cuyo título se encuentre alfabéticamente antes.



The screenshot shows an Excel spreadsheet with the following data:

	A	B	C	D	E	F	G	H	I
1	BUSCARV: búsqueda exacta								
2									
3									
4									
5		Película	Año	Recaudación (millones \$)			Película	Año	Recaudación
6		Psicosis	1960	60			Los pájaros	1963	11,4
7		Vértigo	1958	7					
8		La ventana Indiscreta	1954	37					
9		Los pájaros	1963	11,4					
10		Con la muerte en los talones	1959	9,8					
11		El hombre que sabía demasiado	1956	11,3					
12									
13									
14									


=BUSCARV(G6;B6:D11;2;0)

En la mayoría de fórmulas puedes sustituir la palabra VERDADERO con un 1, y FALSO con un 0. Así es como se muestra en la imagen.

Ejemplo 2: coincidencia aproximada

En los casos que no requieras obtener una coincidencia exacta puedes utilizar el modo de coincidencia aproximada. Un ejemplo claro se encuentra en el cálculo de comisiones para vendedores.

En el siguiente ejemplo se busca una tasa de comisión en la tabla H5:I9. Los valores de búsqueda provienen de la columna B. Como puedes ver en la imagen, se usa BUSCARV en modo de coincidencia aproximada porque en la mayoría de los casos el **valor_buscado** no se encontrará en la **matriz_buscar_en**.



	A	B	C	D	E	F	G	H	I
1	BUSCARV: búsqueda exacta								
2									
3									
4		Vendedor	Ventas	Comision	Importe comisión			Ventas	Comisión
5		Adrián	30681	9%	2761,29			16000	5%
6		Alejo	31166	10%	3116,6			22000	7%
7		Urano	25255	7%	1767,85			27000	9%
8		Patricia	30056	9%	2705,04			31000	10%
9		Katia	29323	9%	2639,07			35000	11%
10		Rocío	34333	10%	3433,3				
11		Fabiola	24904	7%	1743,28				
12		Eduardo	19483	5%	974,15				
13		Eleonor	33358	10%	3335,8				
14		Máximo	18106	5%	905,3				
15									
16									
17									

=BUSCARV(C5;\$H\$5:\$I\$9;2;1)

BUSCARV escanea los valores de la columna H. Si encuentra una coincidencia exacta, devuelve el valor correspondiente. Si no la encuentra “retrocede” para devolver el valor anterior.

Muy importante: Para que BUSCARV funcione correctamente en modo de coincidencia aproximada, la **matriz_buscar_en** debe estar ordenada ascendentemente por el valor de búsqueda. Si no está ordenada, la función devolverá un resultado erróneo.

BUSCARV y los errores #N/D

Al trabajar con BUSCARV es inevitable encontrarse con errores #N/D (#N/A en algunas versiones de Excel). Ver en una celda este mensaje realmente no significa que se trate de un error. El verdadero significado de #N/D o #N/A es “No disponible” o “Not available”.

En la práctica hay varias razones por las que se muestra. Entre ellas destacan:

- Se ha buscado un valor que no existe en la tabla.
- El **valor_buscado** está mal escrito o contiene espacios delante o detrás.
- El rango de la **matriz_buscar_en** no se ha introducido correctamente.

Más adelante se mostrarán más ejemplos prácticos con BUSCARV.

BUSCARH

BUSCARH es una función muy parecida a la anterior. La diferencia es que está diseñada para realizar búsquedas horizontales. Más concretamente, busca un valor en la primera fila de una tabla y devuelve otro valor en la misma columna de la fila que especifiques.

Sintaxis de BUSCARH

La función BUSCARH cuenta con los siguientes argumentos:

BUSCARH(valor_buscado; matriz_buscar_en; indicador_filas; [ordenado])

- **Valor_buscado:** El valor a buscar. Puede ser una referencia de celda, un valor numérico o una cadena de texto.
- **Matriz_buscar_en:** El rango o la tabla donde se va a buscar **valor_buscado**. El rango puede tener definido un nombre. La primera fila siempre debe contener los valores de búsqueda.
- **Indicador_filas:** El número de fila que devolverá la función.
- **Ordenado:** Valor lógico que indica a BUSCARH si debe funcionar en el modo de coincidencia exacta o aproximada. Si se establece como VERDADERO o no se indica, se activa el modo de coincidencia aproximada. Esto significa que si no se encuentra el **valor_buscado** la función devolverá el valor más alto que sea menor que el **valor_buscado**. Si el argumento es FALSO, la función solo devolverá la coincidencia exacta. Si el valor no se encuentra, BUSCARH devuelve un error #N/D. Recuerda que para realizar búsquedas con este modo, la **matriz_buscar_en** debe estar ordenada por la fila de búsqueda.

Para comprobar el funcionamiento de BUSCARH, supón que tienes una tabla con información sobre los planetas del Sistema Solar y deseas obtener el periodo de rotación del planeta que se encuentra en la celda superior.

	A	B	C	D	E	F	G	H	I	J
1	BUSCAR: ejemplo básico									
2										
3										
4		Planeta	Mercurio	Venus	Tierra	Marte	Júpiter	Saturno	Urano	Neptuno
5		Distancia (km.)	57.909.175	108.208.930	149.597.870	227.936.640	778.412.010	1.426.725.400	2.870.972.200	4.498.252.900
6		Rotación (días)	58,646225	243,0187	0,99726968	1,02595675	0,41354	0,44401	0,71833	0,67125
7		Densidad	5,43	5,24	5,515	3,94	1,33	0,697	1,29	1,76
8										
9										
10										
11		Planeta	Marte							
12		Rotación	¿?							
13										
14										
15										

En la fórmula se utilizarán los siguiente argumentos:

- **Valor_buscado:** C11 es la celda que contiene el valor a buscar.
- **Matriz_buscar_en:** C4:J7 es el rango de celdas donde se buscará el valor.
- **Indicador_filas:** es la fila 3 contando desde la parte superior de la matriz de búsqueda.
- **Ordenado:** Falso. Si no se encuentra el valor buscado, no será correcto que la función devuelva un valor aproximado. Además, el rango de búsqueda no se encuentra ordenado por la primera fila.

Juntando todos los argumentos en la fórmula, quedaría de la siguiente manera:

=BUSCARH(C11;C4:J7;3;FALSO)

BUSCAR: ejemplo básico										
1										
2										
3										
4		Planeta	Mercurio	Venus	Tierra	Marte	Júpiter	Saturno	Urano	Neptuno
5		Distancia (km.)	57.909.175	108.208.930	149.597.870	227.936.640	778.412.010	1.426.725.400	2.870.972.200	4.498.252.900
6		Rotación (días)	58,646225	243,0187	0,99726968	1,02595675	0,41354	0,44401	0,71833	0,67125
7		Densidad	5,43	5,24	5,51	3,94	1,33	0,697	1,29	1,76
8										
9										
10										
11		Planeta	Marte							
12		Rotación	1,02595675							
13										
14										

Algunas cosas que debes conocer sobre BUSCARH

Siempre que hagas una búsqueda horizontal debes tener en cuenta que:

1. Solo es posible buscar en la fila superior de la **matriz_buscar_en**.
2. No hay distinción entre mayúsculas y minúsculas.
3. Si el argumento **ordenado** se introduce como VERDADERO (o 1), los valores de la primera fila se deben ordenar de forma ascendente. De lo contrario la función devolverá un valor erróneo.

Ejemplos básicos

En la sección anterior te he mostrado cómo utilizar BUSCARH con coincidencia exacta. Aquí vemos un ejemplo con coincidencia aproximada:

Ejemplo 1: coincidencia aproximada

Supón que tienes una lista de planetas similar a la anterior donde se muestran las densidades y los nombres. Lo que deseas buscar es qué planeta tiene una densidad similar al valor introducido.

Imagina que la densidad que buscas es de 4,2.

INDICE+COINCIDIR

Hasta ahora hemos visto varias formas “directas” de hacer búsquedas. Únicamente se ha utilizado una función en cada una. En la mayoría de las ocasiones, las búsquedas reales que tendrás que hacer serán mucho más complejas y las funciones BUSCARX, BUSCARV o BUSCARH posiblemente se queden cortas.

Existen muchas alternativas para hacer búsquedas más flexibles y potentes. La combinación de INDICE y COINCIDIR es una de ellas (también es la más conocida).

Como el objetivo es demostrar el uso combinado de las dos funciones, no voy a entrar en detalles sobre sus sintaxis a nivel individual.

La función INDICE

Devuelve un valor en una matriz en función de un número de columna y un número de fila específicos. Esta función tiene dos sintaxis diferentes (vectorial o matricial). En este caso, la que nos interesa es la matricial:

INDICE(matriz; núm_fila; [núm_columna])

- **Matriz:** Es el rango de celdas donde se encuentra el valor a devolver.
- **Núm_fila:** El número de fila en la matriz desde la que se va a devolver el valor.
- **Núm_columna:** El número de columna de la matriz desde la que se va a devolver el valor. Es un argumento opcional.

Supón que cuentas con la tabla de la siguiente imagen. La función INDICE devuelve el valor que se encuentra en una determinada fila de una determinada columna:

=INDICE(D5:H9;K5;K6)

	A	B	C	D	E	F	G	H	I	J	K
1											
2											
3											
4				1	2	3	4	5			
5			1	A	F	K	P	U			2
6			2	B	G	L	Q	V		Columna	3
7			3	C	H	M	R	W		Resultado	L
8			4	D	I	N	S	X			
9			5	E	J	O	T	Y			
10											
11											

La función COINCIDIR


Por su parte, la función COINCIDIR busca un valor en un rango de celdas y devuelve su posición relativa en ese rango. Observa su sintaxis:

COINCIDIR(valor_buscado;matriz_buscada; [tipo_de_coincidencia])

- **Valor_buscado:** Es el valor que se desea buscar.
- **Matriz_buscada:** indica el rango de celdas donde se hará la búsqueda.
- **Tipo_de_coincidencia:** Especifica si se debe devolver una coincidencia exacta o aproximada:
 - **1:** Busca el valor más alto que es menor o igual que el **valor_buscado**. Es necesario ordenar la matriz de forma ascendente. Es el valor predeterminado en caso de omisión.
 - **0:** Busca el primer valor exacto en la **matriz_buscada**. Cuando COINCIDIR se usa de forma conjunta con la función INDICE, suele ser el valor a introducir.
 - **-1:** Busca el valor más bajo que es mayor o igual que el **valor_buscado**. Es necesario ordenar la matriz de forma descendente.

Supón que tienes la siguiente lista de países y estás buscando la posición en la que se encuentra Perú dentro de ella:

=COINCIDIR(E4;B5:B11;0)



The screenshot shows an Excel spreadsheet with the following data:

	A	B	C	D	E	F
1						
2						
3						
4		País		País	Perú	
5		Italia		Posición	3	
6		Alemania				
7		Perú				
8		Colombia				
9		Francia				
10		México				
11		Japón				
12						
13						

The formula bar shows the formula: =COINCIDIR(E4;B5:B11;0)

Búsquedas combinando INDICE y COINCIDIR

Ahora que conoces la utilidad de ambas funciones, seguro que te has dado cuenta del tipo de datos que devuelve cada una. COINCIDIR devuelve valores numéricos e INDICE devuelve valores de texto.

INDICE devuelve un valor en función de los números de fila y de columna, y COINCIDIR le proporciona esos números.

Supón que tienes un listado de países con sus respectivas monedas y códigos ISO. Para encontrar el código ISO de una determinada moneda, por ejemplo, la lempira de Honduras, se utiliza la siguiente fórmula:

=INDICE(D6:D25;COINCIDIR(G5;C6:C25;0))



The screenshot shows an Excel spreadsheet with the following data:

	A	B	C	D	E	F	G
1	Funciones INDICE+COINCIDIR						
2							
3							
4							
5							
6		Pais	Moneda actual	Código ISO		Moneda	Lempira
7		Argentina	Peso argentino	ARS		Código	HNL
8		Bolivia	Boliviano	BOB			
9		Brasil	Real brasileño	BRL			
10		Chile	Peso chileno	CLP			
11		Colombia	Peso colombiano	COP			
12		Costa Rica	Colón costarricense	CRC			
13		Cuba	Peso cubano	CUP			
14		Ecuador	Dólar estadounidense	USD			
15		El Salvador	Dólar estadounidense	SVC			
16		Guatemala	Quetzal	GTQ			
17		Haití	Gourde haitiano	HTG			
18		Honduras	Lempira	HNL			

Analicemos con detalle la fórmula anterior:

- La función COINCIDIR busca el valor de la celda G5 (Lempira) en el rango C6:C25 y devuelve 12 porque “Lempira” ocupa la duodécima posición en la matriz de búsqueda.
- El número de fila va directamente al argumento **Núm_fila** de la función INDICE, que le indica que devuelva el valor que se encuentra en esa fila.

Importante: El número de filas en el argumento **matriz** de INDICE debe coincidir con el número de filas del argumento **matriz_buscada** de COINCIDIR.

Este ejemplo se puede resolver más fácilmente con la función BUSCARV. Se ha creado únicamente para demostrar el funcionamiento conjunto de INDICE + COINCIDIR.

Funciones de búsqueda y tablas

En los ejemplos anteriores hemos trabajado con rangos de celdas. Cuando se realiza una búsqueda, todos los rangos utilizados tienen características similares:

- Tienen una columna de búsqueda (que suele ser la primera).
- Tienen una o varias columnas que contienen los datos que se devolverán.
- Pueden aumentar o disminuir las filas columnas con el paso del tiempo (se pueden agregar nuevos clientes o productos).

Este último punto puede suponer un problema cuando se trabaja con rangos delimitados por referencias (por ejemplo B5:D17) porque:

- Cuando añades un nuevo cliente o producto, debes modificar la fórmula para que la matriz de búsqueda incluya los nuevos datos.
- Cada vez que agregas una fila o columna te tienes que asegurar de que no dejas fuera del rango filas o columnas donde puedan encontrarse datos que necesites extraer.

Para despreocuparte de esto existe una solución muy sencilla: utilizar tablas.

Cuando se usan tablas al realizar búsquedas de datos es posible utilizar nombres y referencias estructuradas, evitando estar constantemente haciendo modificaciones en la fórmula cuando se agregan nuevos registros o campos.

Cómo crear una tabla de Excel

La forma más rápida de crear una tabla es situarse en cualquier parte del rango que desees convertir y presionar **Ctrl + T**. También puedes hacerlo desde la ficha **Inicio > Estilos > Dar formato como tabla**.

Comercial	Nº clientes	Ventas netas	Comisiones
Manuel	8	1592	562,772
Alfonso	8	1088	396,9024
Eduardo	8	1680	752,64
María	9	2133	922,7358
Adriana	10	1610	579,117
Javier	10	1540	569,8
Adolfo	7	1316	427,5684
Sergio	7	1799	708,806
Juan Antonio	8	1624	621,3424
Pedro	6	726	235,8774
Guillermo	9	2277	965,6757
Laura	6	714	220,983
Silvia	9	2682	1023,183



Comercial	Nº cliente	ventas netas	Comision
Manuel	8	1.592	563
Alfonso	8	1.088	397
Eduardo	8	1.680	753
María	9	2.133	923
Adriana	10	1.610	579
Javier	10	1.540	570
Adolfo	7	1.316	428
Sergio	7	1.799	709
Juan Antonio	8	1.624	621
Pedro	6	726	236
Guillermo	9	2.277	966
Laura	6	714	221
Silvia	9	2.682	1.023

Así de fácil, ¡ya tienes creada tu tabla!

¿Qué ocurre al crear una tabla?

Al crear una tabla suceden varias cosas con los datos que contiene. Lo que más destaca es el cambio de formato, lo que permite una mejor visualización.

Como ves, las filas se somborean para facilitar la lectura de los datos. Además, aparecen unos iconos en los encabezados de cada columna. Haciendo clic sobre ellos encontrarás muchas herramientas para trabajar con los campos.

Pero lo más increíble al crear una tabla es que podrás hacer referencia a los datos usando referencias estructuradas. Se trata de una nomenclatura de datos que facilita cualquier acción sobre ellos.

Por ejemplo, si quieres hacer referencia a la columna “Nº clientes”, ya no tienes que recordar el rango donde se ubica (F7:F19). Ahora puedes escribir el nombre del campo de la siguiente forma:

Tabla1[Nº clientes]

“Tabla1” hace referencia al nombre de la tabla y “Nº clientes” a la columna que contiene el número de clientes. Antes de utilizar las referencias estructuradas es conveniente dar un nombre a la tabla. Para ello, selecciona cualquier celda de la tabla y pulsa en la ficha contextual **Diseño de tabla**.

Comercial	Nº cliente	Ventas net.	Comision
Manuel	8	1.592	563
Alfonso	8	1.088	397
Eduardo	8	1.680	753
María	9	2.133	923
Adriana	10	1.610	579
Javier	10	1.540	570
Adolfo	7	1.316	428
Sergio	7	1.799	709
Juan Antonio	8	1.624	621
Pedro	6	726	236

Procura dar a la tabla un nombre que describa los datos que contiene, de esta forma será más fácil recordar el nombre en caso de que trabajes con varias tablas a la vez.

Los nombres de las tablas deben ser únicos en cada libro de Excel.

En la imagen anterior puedes comprobar cómo he cambiado el nombre genérico “Tabla1” por “Ventas”. De esta forma ya puedo utilizar ese nombre en cualquier parte del libro:

- Para sumar el total de las ventas puedo usar la fórmula =SUMA(Ventas[Ventas netas])
- Para calcular el promedio de comisión obtenida puedo utilizar =PROMEDIO(Ventas[Comisiones])

Sobre las referencias estructuradas

Las referencias estructuradas utilizan los nombres de las tablas y los campos para referirse a determinados datos. Son muy útiles para hacer las fórmulas más fáciles de entender, pues utilizan un lenguaje más natural que una referencia de celda.

Como el objetivo del ebook es enseñarte a dominar las principales funciones para buscar datos en Excel, no voy a profundizar en la explicación de las referencias estructuradas, pero te animo a que investigues un poco más acerca de ellas, pues son muy interesantes y te facilitarán el trabajo.

Antes	Después
=BUSCARV("María";\$E\$7:\$H\$19;3;FALSO)	=BUSCARV("María";Ventas;3;FALSO)
=BUSCARV("Adolfo";\$E\$7:\$H\$19;2;FALSO)	=BUSCARV("Adolfo";Ventas;2;FALSO)
=BUSCARV("Pedro";\$E\$7:\$H\$19;4;FALSO)	=BUSCARV("Pedro";Ventas;4;FALSO)
=BUSCARV("Joaquín";\$E\$7:\$H\$19;2;FALSO)	=BUSCARV("Joaquín";Ventas;2;FALSO)

¿Por qué usar referencias estructuradas en lugar de referencias de rango?

Seguro que te estás haciendo esta pregunta. En la tabla anterior únicamente se ha modificado la referencia E7:H19 por el nombre de la tabla y apenas hay diferencia, pero existen varias ventajas al hacerlo:

1. Nunca más tendrás que recordar la referencia que ocupa el rango de celdas. Es mucho más fácil recordar la palabra “Ventas” que una referencia como \$E\$7:\$H:\$19.
2. Al agregar más datos a la tabla, ésta se amplía automáticamente aumentando también el rango al que se refiere. De esta forma, no tendrás que hacer modificaciones en las fórmulas.
3. Solo tendrás que introducir la fórmula una vez. El resto de celdas de la columna se autorellenará al presionar **Intro**.

Otras utilidades de las tablas

El acceso a los datos mediante referencias estructuradas es muy útil cuando se utilizan varias funciones (por ejemplo INDICE y COINCIDIR).

En cuanto a la gestión de datos, las tablas cuentan con herramientas accesibles desde el desplegable de los encabezados de los campos. Gracias a estas herramientas es posible ordenar los datos y filtrarlos con un sinfín de criterios.

Otra utilidad de una tabla es que ofrece la posibilidad de mostrar filas y columnas de totales. Estos totales no se limitan a sumar los datos que hay por encima o a la izquierda sino también a hacer otros cálculos como promedios, mostrar los valores máximos o mínimos o contar el número de registros que tiene la tabla.

Uso de comodines

En Excel se llaman “comodines” a unos caracteres especiales que permiten buscar parte de un valor o un número de caracteres específico. Te lo muestro con un ejemplo.

Observa los datos de la siguiente imagen. Puede que no recuerdes el nombre completo de los comerciales de tu empresa, pero sabes que el nombre de uno de ellos comienza por “Jav”. ¿Cómo puedes conocer las ventas de ese comercial si no sabes el nombre completo y por tanto no puedes hacer una búsqueda convencional?

Uso de comodines en las funciones de búsqueda				
	Comercial	Nº clientes	Ventas netas	Comisiones
5	Manuel	8	1592	562,772
6	Alfonso	8	1088	396,9024
7	Eduardo	8	1680	752,64
8	María	9	2133	922,7358
9	Adriana	10	1610	579,117
10	Javier	10	1540	569,8
11	Adolfo	7	1316	427,5684
12	Sergio	7	1799	708,806
13	Juan Antonio	8	1624	621,3424
14	Pedro	6	726	235,8774
15	Guillermo	9	2277	965,6757
16	Laura	6	714	220,983
17	Silvia	9	2682	1023,183

La solución es sencilla. Tendrás que utilizar el comodín “*” para encontrarlo. Fíjate en cualquiera de las fórmulas:

```
=BUSCARX("Jav*";B5:B17;D5:D17;;2)
```

```
=BUSCARV("Jav*";B5:E17;3;0)
```

```
=INDICE(B5:E17;COINCIDIR("Jav*";B5:B17;0);3)
```

Con el comodín “*” le estás diciendo a Excel que pueden existir cualquier número de caracteres detrás del valor buscado. Lo que realmente importa es que comience con “Jav”.

Te muestro más ejemplos con comodines:

Dato a hallar	Fórmula
Ventas del comercial cuyo nombre tiene 12 caracteres	=BUSCARV("????????????";\$B\$5:\$E\$17;3;FALSO)
Ventas del comercial cuyo nombre termina por er	=BUSCARV("*er";\$B\$5:\$E\$17;3;FALSO)
Clientes del comercial cuyo nombre comienza por G y termina por o	=BUSCARV("G*o";\$B\$5:\$E\$17;2;FALSO)

Existen tres caracteres comodín:

- *: Sustituye cualquier número de caracteres.
- ?: Sustituye solo una posición.
- ~: Únicamente se utiliza cuando quieres hacer búsquedas de asteriscos o interrogaciones.

Atención: Acabamos de ver que los comodines son muy útiles en cierto tipo de búsquedas, pero debes tener precaución con el resultado que devuelven, pues siempre muestran la primera coincidencia que encuentran. Es decir, buscas un nombre que comienza por "Jua", y en el listado hay una persona que se llama Juan, otra que se llama Juan Antonio y otra que se llama Juana, debes tener en cuenta que la búsqueda sólo devolverá el nombre que encuentre primero.

Además de las funciones BUSCARX, BUSCARV, INDICE y COINCIDIR, existen más funciones que admiten caracteres comodín. Las más usadas son SUMAR.SI, ENCONTRAR, HALLAR o REEMPLAZAR.

Más adelante veremos algunos ejemplos de uso de comodines.

Validación de datos con funciones de búsqueda

Si llevas trabajando con Excel algún tiempo te habrás dado cuenta de que el verdadero poder de Excel no se encuentra en funciones como **BUSCARX**, **BUSCARV** o en otras funciones, ni siquiera en las tablas dinámicas o los formatos condicionales, sino en la capacidad de combinar dos o más características para conseguir el objetivo deseado.

Por ejemplo, puedes utilizar formatos condicionales dentro de una tabla para que, al introducir nuevos datos, se apliquen automáticamente.

En esta sección te mostraré cómo combinar BUSCARX o BUSCARV con validación de datos para crear una pequeña aplicación de facturación.

¡Adelante!

[Pedro, el empleado con el trabajo más aburrido del mundo](#)

Te presento a Pedro, un empleado que trabaja en el departamento de facturación de una empresa multinacional. Una de sus tareas consiste en analizar miles de facturas todos los meses para dar soporte telefónico a los clientes que le llaman preguntando por ellas.

Lo peor del trabajo de Pedro es lo que él llama las *“llamadas de los estados”*. Todos los días debe contestar muchas llamadas de clientes que le preguntan en qué estado de pago se encuentran sus facturas. Pedro, con su listado de facturas, se limita a filtrar una tabla por el número de factura y a responder al cliente con el dato que necesita.

Un día, aburrido por su trabajo tan monótono, decide ponerse en tus manos para que le eches una mano. Pedro confía en ti, ya que sabes mucho más de Excel que él... y no le puedes defraudar.

Te dispones a ayudarlo:

[¿Cómo están estructurados los datos de Pedro?](#)

Pedro tiene una tabla de datos como esta:

Nº factura	Fecha	Importe	Emitida a	País de envío	Descripción	Notas	¿Pagada?	Fecha pago
FV00001	09/10/2013	24600	Gamify dot com	Argentina	Pagar factura inmediatamente	Notas de facturación	No	
FV00002	09/09/2013	12200	Bang Inc.	Argentina	Pagar factura en 1 mes	Notas de facturación	No	
FV00003	11/11/2013	4900	Elephant Design	Austria	Pagar factura en 1 mes	Notas de facturación	No	
FV00004	01/11/2013	17700	Fancy Family Biz	Finlandia	Pagar factura inmediatamente	Notas de facturación	No	
FV00005	06/03/2014	5500	Bang Inc.	España	Pagar factura inmediatamente	Notas de facturación	No	
FV00006	10/02/2014	22400	Helberth Associates	China	Pagar factura inmediatamente	Notas de facturación	No	
FV00007	02/10/2013	24400	Helberth Associates	Canada	Pagar factura en 1 semana	Notas de facturación	No	
FV00008	20/12/2013	26400	Divided Analysts	Uruguay	Pagar factura en 1 semana	Notas de facturación	No	
FV00009	08/11/2013	15100	Gamify dot com	Canada	Pagar factura inmediatamente	Notas de facturación	No	
FV00010	19/02/2014	20000	Abbot Co.	Colombia	Pagar factura tras hablar con Abbot Co.	Notas de facturación	No	
FV00011	06/11/2013	15600	Abbot Co.	Chile	Pagar factura inmediatamente	Notas de facturación	No	
FV00012	11/01/2014	7400	Fancy Family Biz	Argentina	Pagar factura inmediatamente	Notas de facturación	No	
FV00013	02/10/2013	9000	Elephant Design	China	Pagar factura en 1 mes	Notas de facturación	No	
FV00014	07/01/2014	26200	Helberth Associates	Uruguay	Pagar factura tras hablar con Helberth Associates	Notas de facturación	No	
FV00015	14/11/2013	28800	Bang Inc.	Argentina	No pagar factura	Notas de facturación	No	
FV00016	31/12/2013	22800	Fancy Family Biz	China	Pagar factura tras hablar con Fancy Family Biz	Notas de facturación	No	
FV00017	17/12/2013	13500	Elephant Design	Chile	Pagar factura en 1 mes	Notas de facturación	No	
FV00018	27/11/2013	20500	Cater Data Corp	Colombia	Pagar factura en 1 semana	Notas de facturación	No	
FV00019	10/02/2014	14500	Cater Data Corp	Finlandia	Pagar factura en 1 mes	Notas de facturación	No	
FV00020	06/02/2014	20300	Gamify dot com	Francia	Pagar factura en 1 semana	Notas de facturación	No	
FV00021	03/12/2013	25100	Gamify dot com	China	Pagar factura inmediatamente	Notas de facturación	No	
FV00022	17/02/2014	8300	Indigo Decors	Argentina	Pagar factura en 1 mes	Notas de facturación	No	
FV00023	25/10/2013	13400	Bang Inc.	Uruguay	No pagar factura	Notas de facturación	No	
FV00024	02/02/2014	17000	Divided Analysts	Chile	No pagar factura	Notas de facturación	No	
FV00025	08/12/2013	26500	Elephant Design	Colombia	Pagar factura en 1 mes	Notas de facturación	No	
FV00026	20/02/2014	6400	Elephant Design	Canada	Pagar factura tras hablar con Elephant Design	Notas de facturación	No	
FV00027	28/10/2013	28800	Helberth Associates	España	Pagar factura inmediatamente	Notas de facturación	No	
FV00028	19/12/2013	25400	Divided Analysts	España	Pagar factura inmediatamente	Notas de facturación	No	
FV00029	09/10/2013	24600	Gamify dot com	Argentina	Pagar factura inmediatamente	Notas de facturación	No	

Como puedes comprobar, la primera columna contiene el número de factura. Las demás columnas contienen datos como la fecha, el importe o el estado de pago en el que se encuentran.

Este es el proceso que a Pedro le gustaría seguir para desarrollar su trabajo sin problema:

1. Recibir una *“llamada de los estados”*.
2. Escribir el número de factura en una celda.
3. Pulsar **Intro** para que aparezcan los detalles de la factura en la pantalla.
4. Indicar esos detalles a su interlocutor.

Para llegar al objetivo de Pedro, primero tienes que crear una plantilla donde se ubiquen todas las fórmulas que devolverán los datos buscados. Algo como esto:

Datos de factura: FV00012


Fecha:	11/01/2014	Importe	7.400,00
Emitida a:	Fancy Family Biz		
Enviada a:	Argentina		
Descripción	Pagar factura inmediatamente		
Notas:	Notas de facturación		
¿Pagada?	No	Fecha de pag	

Una vez que tienes la tabla con los datos y la plantilla con el informe de visualización, es hora de ponerse a trabajar.

Si el listado de facturas no se encuentra en una tabla, créala como te mostré anteriormente. Llámala “Facturas”, que es el nombre más descriptivo que podrías asignarle. Una vez hecho esto, ya es posible referirte a todo el conjunto de facturas como [#Facturas].


Para que Pedro pueda seleccionar un número de factura del cuadro desplegable es necesario crear un nombre que englobe todas las facturas.

1. Selecciona los datos de la columna **B** en la hoja **Datos de facturación**.
2. En la ficha **Fórmulas > Administrador de nombres**, haz clic en **Nuevo**.
3. Utiliza “Nº_Factura” como nombre de rango.
4. En el campo “Se refiere a” asegúrate de que aparece escrita la referencia a la tabla de los datos que has seleccionado en el paso 1. En caso de que no aparezca, puedes hacer clic en el botón situado a la derecha y seleccionar el rango con el ratón.




Ahora que tienes el nombre creado, debes seleccionar la celda que va a contener los números de factura y dirigirte a la ficha **Datos > Validación de datos** para mostrar el cuadro de diálogo **Validación de datos**.

1. Una vez abierto el cuadro de diálogo, selecciona **Lista** en la pestaña **Configuración**.
2. Asegúrate de que la casilla **Celda con lista desplegable** está marcada.
3. En el cuadro **Origen** escribe “Nº_Factura”.



4. Ahora ya es posible seleccionar el número de factura del cuadro desplegable:



Después de conseguir que Pedro pueda seleccionar su factura en la celda E6, debes escribir las fórmulas que conectarán esa celda con los datos de la tabla.

Para mostrar la fecha de la factura seleccionada, sigue estos pasos:

1. Sitúate en la celda E11 y escribe cualquiera de las siguientes fórmulas:
 - a. =BUSCARX(E6;Facturas[Nº factura];Facturas[Fecha])
 - b. =BUSCARV(E6;Facturas;2;0)

Con estas fórmulas le estás indicando a Excel que busque el valor que se encuentra en la celda E6 dentro de la tabla **Facturas**. En la primera fórmula se le indica el campo que debe devolver y en la segunda fórmula, el número de columna. Con BUSCARV es necesario establecer el modo de búsqueda exacto (escribiendo "0" o "FALSO").

2. Siguiendo el ejemplo del paso anterior, y observando los ejemplos de capítulos anteriores, seguro que eres capaz de crear el resto de las fórmulas del documento. Según tu versión de Office, utiliza una de las dos fórmulas propuestas.

De esta forma tan sencilla acabas de crear una pequeña aplicación para que Pedro trabaje de una forma más rápida y eficiente.

Las funciones ENCONTRAR y HALLAR

Hasta ahora hemos experimentado con funciones que realizan búsquedas en rangos de celdas, pero en muchas ocasiones necesitarás buscar ciertos caracteres en una determinada celda para, por ejemplo, conocer la cantidad de caracteres numéricos que tiene la referencia de un producto.

De este tipo de búsquedas “detalladas” se encargan las funciones ENCONTRAR y HALLAR. Aunque son muy similares tienen diferentes usos.

Las vemos a continuación:

La función ENCONTRAR

Se usa para devolver la posición de un carácter o conjunto de caracteres dentro de una cadena de texto.

Su sintaxis es la siguiente:

ENCONTRAR(texto_buscado; dentro_del_texto; [núm_inicial])

- **Texto_buscado.** Es el texto que deseas encontrar.
- **Dentro_del_texto:** Es el texto que, a su vez, contiene el texto que deseas encontrar.
- **Núm_inicial:** Es un argumento opcional. Especifica el número del carácter a partir del cual comienza la búsqueda. El primer carácter es el número 1. Si omites el argumento, se supondrá que quieres buscar desde el primer carácter.

Si el **texto_buscado** no se encuentra en **dentro_del_texto**, la función devuelve el error #¡VALOR!

Por ejemplo, la fórmula =ENCONTRAR(“O”;A2) devuelve 4 porque “O” es la cuarta letra de la palabra “ENCONTRAR”. La fórmula =ENCONTRAR(“I”;A4) devuelve un error porque la letra I no se encuentra en el valor de la celda.

	A	B	C
1	Texto	Resultado	Fórmula
2	ENCONTRAR	4	=ENCONTRAR("O";A2)
3			
4	ENCONTRAR	#¡VALOR!	=ENCONTRAR("I";A4)
5			
6			

Para usar correctamente ENCONTRAR debes tener en cuenta algunas normas:

- Distingue entre mayúsculas y minúsculas. Si no deseas que la función sea sensible a mayúsculas, debes usar la función HALLAR.

- No permite caracteres comodín.
- Si el argumento **texto_buscado** contiene varios caracteres, ENCONTRAR devuelve la posición del primer carácter. Por ejemplo, la fórmula ENCONTRAR("lag";"murciélago") devuelve el valor 7 porque la "l" se encuentra en séptima posición.
- Si el argumento **dentro_del_texto** contiene varias repeticiones de **texto_buscado**, ENCONTRAR devuelve la primera coincidencia. Por ejemplo ENCONTRAR("e";"televisor") devolverá 2 porque es la posición de la primera coincidencia de la letra "e".
- Si **texto_buscado** es una cadena vacía "", la fórmula devuelve el primer carácter de la cadena de búsqueda.
- ENCONTRAR devuelve el error #¡VALOR! si se encuentra en una de estas situaciones:
 - **Texto_buscado** no existe en **dentro_del_texto**.
 - **Núm_inicial** contiene más caracteres que **dentro_del_texto**.
 - **Núm_inicial** es 0 (cero) o un número negativo.

La función HALLAR

La función HALLAR es muy similar a ENCONTRAR, pues sirve para lo mismo. Devuelve la ubicación de uno o varios caracteres dentro de una cadena de texto. También son iguales en su sintaxis:

HALLAR(texto_buscado; dentro_del_texto; [núm_inicial])

La diferencia con la función ENCONTRAR es que HALLAR no distingue entre mayúsculas y minúsculas y permite utilizar caracteres comodín.

Por ejemplo =HALLAR("mercado";"supermercado") devuelve 6 porque la "m" de mercado ocupa la sexta posición dentro del texto.

La fórmula =HALLAR("e";"Excel") devuelve 1 porque e es la primera letra del contenido de la celda ignorando si se trata de mayúscula o minúscula.

Ejemplos prácticos con funciones de búsqueda

Hasta ahora hemos aprendido las sintaxis y los argumentos de las funciones más utilizadas para realizar búsquedas en Excel. También hemos visto algunos ejemplos básicos que te permitirán ponerte en marcha y comenzar a usarlas.

Pero no siempre es fácil. A menudo nos encontraremos con situaciones complejas en las que estos ejemplos no sirvan de mucho, y es necesario agudizar el ingenio para desarrollar fórmulas que cubran nuestras necesidades.

En esta sección se muestran algunos de los ejemplos más habituales de búsquedas con los que espero alcanzar tus expectativas. No obstante, esta lista está abierta a sugerencias. Si consideras que tu búsqueda en particular debería figurar en este listado, escíbeme un correo a sergio@ayudaexcel.com adjuntando tu archivo de Excel con la búsqueda y el máximo detalle posible. En próximas ediciones del manual podrías verla publicada y tu nombre figurando como autor.

BUSCARX con comodines

En Excel se llaman “comodines” a unos caracteres especiales que permiten buscar parte de un valor o un número de caracteres específicos. Un ejemplo vale más que mil explicaciones:

Observa los datos que se muestran en la siguiente imagen. Puede que no recuerdes el nombre completo de los comerciales, pero sabes que el nombre de uno de ellos comienza por “mat”. ¿Cómo puedes extraer los datos de ese comercial si no sabes el nombre completo?

Par poder incluir comodines en las búsquedas, debes introducir el valor 2 en el argumento **modo_de_coincidencia**.

The screenshot shows an Excel spreadsheet with the following data:

ID	Nombre	Apellido	Departamento
1	Napoleón	Barahona	Compras
2	Jose	Caminero	Compras
3	José Luis	Carboné	Dirección
4	Diana	Carboné	Ventas
5	Sagrario	Castelló	Administración
6	Juan José	Dorcas	Calidad
7	Edgardo	Albéniz	Ventas
8	Kurt	Escrivá	Dirección
9	Rosa	Ocampo	Administración
10	Andrés	Casares	Administración
11	Rosalía	Carmona	Administración
12	Matías	Sanchiz	Calidad

The formula bar shows: `=TRANSPONER((BUSCARX(I5;C5:C23;B5:E23;"No encontrado";2)))`

The search criteria in cell I5 is `mat*`. The results are shown in a vertical array in cells J5:J8:

ID	12
Nombre	Matías
Apellido	Sanchiz
Departamento	Calidad

`=TRANSPONER((BUSCARX(I5;C5:C23;B5:E23;"No encontrado";2)))`

En esta fórmula se hace coincidir el valor con comodín que se encuentra en la celda I5 para que devuelva los cuatro campos. La función TRANSPONER es opcional y solo se utiliza para convertir la matriz horizontal en vertical.

¿Cómo funciona la fórmula?

Como todas las fórmulas anidadas, Excel realiza los cálculos de dentro hacia afuera. En primer lugar se evalúa BUSCARX. Como ves, el último argumento contiene el valor 2, lo cual permite buscar valores con comodines.

Como I5 contiene el valor `"mat"`, Excel buscará el primer registro cuyo nombre comience por `"mat"` y devolverá los cuatro valores en una matriz horizontal:

`{12;"Matías";"Sanchiz";"Calidad"}`

Este resultado se introduce como argumento a la función TRANSPONER cuyo objetivo es convertir un rango vertical en horizontal (y viceversa):

`=TRANSPONER({12;"Matías";"Sanchiz";"Calidad"})`

Observa que los separadores de campos antes y después de que TRANSPONER calcule la matriz:

`{12;"Matías";"Sanchiz";"Calidad"}`

Atención: los separadores de campos pueden variar según la configuración regional de tu equipo.

Comodines implícitos

En el ejemplo anterior el comodín se encuentra en la celda, pero en la mayoría de los casos las búsquedas a realizar no serán así. Las celdas contendrán únicamente los valores.

En este caso debemos introducir el comodín en la fórmula de la siguiente manera:

```
=TRANSPONER((BUSCARX(I5&"*";C5:C23;B5:E23;"No encontrado";2)))
```

Como ves se concatena el comodín con el valor de I5 en la propia fórmula. Esto añadirá un asterisco al final de cualquier valor introducido en I5 y BUSCARX realizará la búsqueda incluyéndolo como **valor_buscado**.

Los comodines son muy útiles en cierto tipo de búsquedas pero debes tener precaución con el resultado que devuelven ya que siempre van a devolver la primera coincidencia que se encuentre. Es decir que si haces una búsqueda de la persona cuyo nombre comienza por "Jos", debes tener en cuenta que la búsqueda se detendrá una vez haya encontrado a José, José Antonio o Jose (el valor que encuentre primero).

Si el contenido a buscar puede encontrarse en cualquier parte de la celda (principio, medio o final), debes incluir asteriscos tanto al principio como al final ("**mat**") para que BUSCARX encuentre más coincidencias

Además de BUSCARX o BUSCARV, existen más funciones que admiten caracteres comodín como SUMAR.SI, ENCONTRAR, HALLAR o REEMPLAZAR.

BUSCARX con resultado matricial en diferente orden

Anteriormente vimos que BUSCARX es capaz de devolver varios resultados en forma de matriz. En aquel ejemplo queríamos conocer las calificaciones de un alumno en todas las materias. Estas materias se encontraban ordenadas en la misma posición que el rango de búsqueda.

En algunas ocasiones este orden puede que no sea el adecuado y necesites un resultado matricial personalizado.

BUSCARX puede utilizarse para reordenar columnas anidándola dentro de otra BUSCARX.

```
=BUSCARX(C4:G4;C10:G10;BUSCARX(B5;B11:B32;C11:G32))
```


	A	B	C	D	E	F	G	H	I
1									
2									
3									
4									
5		País	ene	feb	mar	abr		feb	
6		Portugal	308	521	613	524		521	
7		Francia	276	751	607	390		751	
8		Chile	463	938	972	733		938	
9		Alemania	806	985	278	269		985	
10		Uruguay	888	607	665	873		607	
11		Colombia	339	641	528	297		641	
12									
13									
14									
15		Alemania	806	985	278	269			
16									
17									
18									
19									
20									
21									

En el ejemplo de la imagen, la fórmula de H6 es la siguiente:

=BUSCARX(H5;C5:F5;C6:F11)

Como el argumento **matriz_buscada** es un rango horizontal, se supone que la **matriz_devuelta** será el rango en vertical en la misma columna del valor que se buscaba.

BUSCARX sensible a mayúsculas y minúsculas

La mayor parte de las funciones que veremos en este manual no son sensibles a mayúsculas y minúsculas, es decir, que al hacer una búsqueda de la palabra "Excel", podemos encontrarnos con los resultados de las búsquedas "ExCeL", "excel", o "EXCEL".

Para hacer que BUSCARX diferencie entre mayúsculas y minúsculas al hacer la búsqueda debemos incluir la función IGUAL. En el ejemplo mostrado en la imagen, la fórmula de la celda H8 es:

=BUSCARX(1;--IGUAL(B6:B22;G8);C6:E22)

Referencia	ene	feb	mar
aC	398	395	393
aa	382	194	173
AC	271	279	244
Cc	191	108	471
CC	484	465	467
BC	325	406	492
aC	120	254	254
Cb	492	323	312
BC	348	213	110
CC	348	159	314
CB	369	171	318
Cb	234	470	165
cA	403	292	296
BA	134	166	328
Bc	101	274	260
aB	247	180	485
bb	221	207	244

Como ves, la fórmula encuentra la primera coincidencia de "CC" (las dos letras en mayúscula).

La misión de IGUAL es comparar dos valores. El resultado es VERDADERO cuando ambos son iguales y FALSO cuando no lo son. Al introducirla como en la celda H8 el resultado devuelto es una matriz de 17 valores, que compara cada uno de los valores del rango con el valor de comparación. Este es el resultado de la función IGUAL:

{FALSO;FALSO;FALSO;FALSO;VERDADERO;FALSO;FALSO;FALSO;FALSO;VERDADERO;FALSO;FALSO;FALSO;FALSO;FALSO;FALSO;FALSO}

Observa que los valores VERDADERO corresponden con las filas en las que el valor coincide con "CC".

Para hacer más fácil la comprensión de la fórmula y evitar posibles errores con la lógica booleana, los valores VERDADERO y FALSO se convierten en valores 1 y 0, utilizando una doble negación (--) delante de la función IGUAL:

{0;0;0;0;1;0;0;0;0;1;0;0;0;0;0;0;0}

Esta matriz resultante se introduce como **matriz_buscada** en la función BUSCARX

=BUSCARX(1;{0;0;0;0;1;0;0;0;0;1;0;0;0;0;0;0;0};C6:E22)

El **valor_buscado** "1", BUSCARX lo encuentra en la quinta posición, devolviendo la quinta fila en la **matriz_devuelta**.

Meses con mayor y menor venta

En muchas ocasiones el **valor_buscado** depende del valor de la celda. En el ejemplo de la imagen se están buscando los meses en los que hubo mayor y menor venta.

The screenshot shows an Excel spreadsheet with the following data:

Mes	Valor
ene	381
feb	224
mar	154
abr	208
may	379
jun	439
jul	330
ago	288
sep	178
oct	166
nov	426
dic	345

The formula bar shows: **=BUSCARX(MAX(C6:C17);C6:C17;B6:B17)**

The result of the formula is displayed in a green box: **Valor máx. jun** and **Valor mín. mar**. A green arrow points from the formula bar to the result box.

sobre el valor de la columna C (“*Familia Prod.*”), que representa el tipo de producto vendido. Si C6 es “*HM*” entonces se devuelve la primera tabla de búsqueda. Si es otro valor, se devuelve la segunda tabla.

Calcular importe del envío con BUSCARV

Tienes que enviar por correo un paquete y quieres conocer la tasa de envío. Esta tasa se calcula según una tasa fija + un porcentaje del peso del paquete. Observa la imagen:

BUSCARV: buscar en dos tablas según un criterio

Tasa fija		Tasa variable	
Pesos	Tasa	Pesos	Tasa
0	1,2	0	1,3
0,2	1,2	0,2	1,6
0,5	1,2	0,4	2,3
0,8	1,2	0,7	3
1	1,2	1	3,9
1,5	1,5	1,4	5,1
2	1,8	1,9	7
3	2	2,5	8,5
5	2,6	4	10,1

Resultado de la fórmula: **Importe a pagar: 10,3**

En la fórmula =BUSCARV(I5;B5:C13;2;1)+BUSCARV(I5;E5:F13;2;1) puedes comprobar que se realizan dos búsquedas con BUSCARV y se suman sus resultados. Ambas búsquedas son aproximadas, de forma que, si el resultado exacto no se encuentra en la tabla, el valor devuelto es el inmediatamente inferior. Recuerda que para realizar este tipo de búsqueda, las tablas deben estar ordenadas de forma ascendente.

BUSCARV hacia la izquierda

Seguro que has leído en varias ocasiones que BUSCARV únicamente se puede utilizar para realizar búsquedas hacia la derecha. Esto es cierto en parte. Es verdad que BUSCARV por sí sola no puede buscar hacia la izquierda, pero con una pequeña modificación sí es posible. Observa la imagen:

111 X ✓ fx =BUSCARV(\$I\$8;ELEGIR({2\1};\$B\$7:\$B\$83;\$F\$7:\$F\$83);2;0)

	A	B	C	D	E	F	G	H	I
1	BUSCARV: hacia la izquierda								
2									
3									
4									
5									
6		Apellido	Apellido, Nombre	Dirección	Ciudad	Provincia	ID		
7		Lemona Eleizalde, Gabriela	Ebro, 92	Autol	Rioja, La	5		ID	
8		Roa Gonzaga, Koldo	Nemesio Castro, 87	Aceuchal	Badajoz	1			16
9		Novés del Campo, Katyuska	Avenida Monte de Valdelatas,	Porriño, O	Pontevedra	18			
10		Azorín Echábarri, Conrado	Nutria, 34	Aldeatejada	Salamanca	16			
11		Barba Aguado, Constantio	Jazmín, 42	Cañavate, El	Cuenca	6	Nombre	Azorín Echábarri, Conrado	
12		Cardona Sales, Urso	Sueños, 58	Ortigosa de Pestaño	Segovia	15	Dirección	Nutria, 34	
13		Kent Carranza, Verónica	Julián López Silva, 56	Cabañas de Yepes	Toledo	17	Ciudad	Aldeatejada	
14		Navascués Sagarra, Urbana	Santa Isabel, 12	Zarzuela de Jadraque	Guadalajara	12	Provincia	Salamanca	
15		Agramont Cifontes, Rita	Vereda de los Llanos, 15	Idiazabal	Gipuzkoa	13			
16		Abráldex Campana, Angustias	Manuel Serrano Frutos, 10	Yésero	Huesca	8			
17		Niño Ajofrín, Samuel	Pío Baroja, 35	Terriente	Teruel	4			
18		Bécquer Batres, Nuria	Córdoba, 86	Torres de Alcanadre	Huesca	10			
19		Alcocer Lago, Zulema	Luis Rodríguez Ontiveros, 86	Cervera de Pisuerga	Palencia	3			
20		Cabanillas Bouffart, Crisóstomo	Granada,	Cubo de Benavente	Zamora	11			
21		Adarre Grandes, Linda	Cádiz, 85	Ausejo de la Sierra	Soria	2			
22		Escrivá Goiri, Franco	Tajuña, 66	Leaburu	Gipuzkoa	7			
23		Sartorius Alfonsín, Blas	Travesía del Abeto, 38	San Lorenzo de Calatrava	Ciudad Real	9			
24									
25									
26									

En el listado anterior figuran los datos de algunas personas. Como puedes comprobar, la columna ID se encuentra en la parte derecha de la tabla.


BUSCARV es capaz de buscar el número en la columna ID y devolver cualquiera de los otros datos. Como no es posible utilizar directamente BUSCARV, se usa ELEGIR para invertir el orden de las columnas.

Esta es la fórmula utilizada en la celda I11 para que devuelva el nombre de la persona:

En España: =BUSCARV(\$I\$8;ELEGIR({2\1};\$B\$7:\$B\$83;\$F\$7:\$F\$83);2;0)

En América Latina y países anglosajones:

=BUSCARV(\$I\$8;ELEGIR({2,1};\$B\$7:\$B\$83;\$F\$7:\$F\$83),2,0)



- **Dato a buscar:** Es el dato que quieres buscar y desde el cual se comenzará a contar el número de columna de búsqueda.
- **Se invierte el orden de las columnas:** Aquí está la clave de la fórmula. La matriz (encerrada entre llaves) le dice al resto de argumentos de la función ELEGIR, cuál va en primer lugar y cuál en segundo.
- **Primera columna y segunda columna:** Son rangos de celdas. En la primera se encuentra el dato que se va a devolver y en la segunda el dato buscado.

- **Columna de búsqueda:** Es el número de columna donde se encuentra el dato a devolver.
- **Ordenado:** Para hacer una búsqueda exacta o aproximada.

Como te vas a dar cuenta ahora, la columna de búsqueda siempre va a ser 2, ya que solo hay dos columnas que intervienen en la fórmula (la de búsqueda y la de resultado).

Para verlo más claro, te pongo otro ejemplo:

¿Cómo sería la fórmula si quisieras buscar la dirección de una de las personas sabiendo su ID?

Únicamente tendrías que sustituir el rango de la primera columna.

`=BUSCARV(I8;ELEGIR({2\1});C7:C83;F7:F83);2;0)`

La única diferencia que existe sobre la fórmula anterior es la referencia a la columna de resultados. Tanto la columna de búsqueda como el número de columna siguen siendo los mismos.

BUSCARV en otra hoja u otro libro

Hasta ahora, hemos hecho búsquedas en tablas de datos que se encontraban en la misma hoja. Para buscar datos de productos, precios u otra información almacenada en una hoja diferente u otro libro independiente, podemos utilizar BUSCARV introduciendo la referencia completa a los datos, ya sea en otra hoja del mismo libro o en un libro que se encuentre en otra carpeta del equipo, e incluso otros equipos conectados en red.

	A	B	C	D	E	F	G	H
1	BUSCARV: en otras ubicaciones							
2								
3								
4								
5			ID	16				
6								
7								
8		En otra hoja	<code>=BUSCARV(D5;'Datos ejemplo'!C7:G23;2;0)</code>					
9		En otro libro de la misma carpeta	<code>=BUSCARV(D5;'[Otro libro.xlsx]Datos ejemplo'!C7:G23;2;0)</code>					
10		En un libro de otra carpeta	<code>=BUSCARV(D5;'C:\Users\Usuario\Documents\Otro libro.xlsx]Datos ejemplo'!C7:G23;2;0)</code>					
11		En otro equipo en red	<code>=BUSCARV(D5;'\\Servidor\carpeta\[Otro libro.xlsx]Datos ejemplo'!C7:G23;2;0)</code>					
12								
13								
14								
15								
16								
17								
18								

Todos los ejemplos de la imagen anterior buscan el valor de la celda D5:

- En otra hoja: =BUSCARV(D5;'Datos ejemplo'!C7:G23;2;0)
- En otro libro de la misma carpeta: =BUSCARV(D5;'[Otro libro.xlsx]Datos ejemplo'!C7:G23;2;0)
- En un libro de otra carpeta:
=BUSCARV(D5;'C:\Users\Usuario\Documentos\[Otro libro.xlsx]Datos ejemplo'!C7:G23;2;0)
- En otro equipo ubicado en la misma red:
=BUSCARV(D5;'\\Servidor\carpeta\[Otro libro.xlsx]Datos ejemplo'!C7:G23;2;0)

Como puedes observar las referencias a los diferentes libros están encerradas entre corchetes y las rutas y las hojas entre apóstrofes. Estos apóstrofes son necesarios cuando el nombre del libro o la hoja contienen espacios o caracteres de puntuación.

BUSCARV en tablas de doble entrada

Dentro de la función BUSCARV hemos visto que el argumento **Indicador_columnas** normalmente suele ser un número estático. Sin embargo, podemos crear un indicador de columnas dinámico utilizando la función COINCIDIR para localizar la columna adecuada. Esta técnica permite crear una búsqueda bidireccional dinámica emparejando tanto las filas como las columnas. También puede hacer que la función BUSCARV sea más consistente. Por desgracia BUSCARV puede “romperse” cuando se insertan o eliminan columnas en una tabla. La combinación de funciones BUSCARV + COINCIDIR evita este problema incluso si se hacen cambios en las columnas.

=BUSCARV(I5;B6:F11;COINCIDIR(I6;C5:F5;0)+1;0)									
	A	B	C	D	E	F	G	H	I
1	BUSCARV+COINCIDIR: búsquedas más consistentes								
2									
3									
4									
5		País	ene	feb	mar	abr		País	Uruguay
6		Portugal	308	521	613	524		Mes	feb
7		Francia	276	751	607	390			
8		Chile	463	938	972	733			
9		Alemania	806	985	278	269			
10		Uruguay	888	607	665	873		Resultado	607
11		Colombia	339	641	528	297			
12									
13									
14									

=BUSCARV(I5;B6:F11;COINCIDIR(I6;C5:F5;0)+1;0)

I5 proporciona el valor de búsqueda para la fila y I6 para la columna.

Como resultado, la fórmula de D6 obtiene el nombre de la tabla de clientes (columna 2), y la fórmula de E6 obtiene la dirección (columna 3).

Es una excelente forma de copiar varias columnas para recuperar valores de columnas consecutivas en otra tabla.

BUSCARH para encontrar el valor de la primera celda no vacía

Para buscar y obtener el primer valor en un rango horizontal, puedes utilizar BUSCARH con un comodín. Observa el ejemplo.

	A	B	C	D	E	F	G	H
1	BUSCARV: fusionar tablas							
2								
3								
4								
5								
6		Departamento	Trim.1	Trim.2	Trim.3	Trim.4		1ª aparición
7		Compras	A					A
8		Administración		A				A
9		Dirección			C			C
10		Calidad		B				B
11		Diseño		C				C
12		Marketing	C		D			C
13		Publicidad		B	C	A		B
14		Formación				C		C
15								
16								
17								
18								
19								

En la fórmula que se muestra en la imagen:

=BUSCARH("*";C7:F7;1;0)

El valor_buscado "*" es el comodín que puede coincidir con uno o más valores de texto.

La **matriz_buscar_en** (C7:F7) se introduce como referencia relativa para que cambie a medida que la fórmula se copie en las demás filas de la columna H.

El argumento **indicador_fila** es 1, pues el rango contiene solo una fila.

El argumento **ordenado** se establece en cero (FALSO) para forzar la coincidencia exacta. Esto es necesario cuando se usan comodines con BUSCARV y BUSCARH.

En cada fila BUSCARH busca y devuelve el primer valor de texto que se encuentra entre las columnas C y F de cada fila.

Mostrar resultados de búsqueda por categoría de forma vertical

Una de las limitaciones de BUSCARV es que no puede devolver varios resultados. Esta tarea puede llevarse a cabo utilizando las siguientes funciones de forma matricial:

- SI: Evalúa una condición y devuelve un valor si se cumple. En caso contrario devuelve otro valor.
- K.ESIMO.MENOR: Devuelve el valor k-ésimo más pequeño de la matriz.
- INDICE: Devuelve un elemento de la matriz en función de los números de fila y columna especificados.
- FILA: Devuelve el número de fila.
- COLUMNA: Devuelve el número de columna.
- SI.ERROR: En caso de que se produzca un error, muestra un mensaje personalizado.

Trabajaremos con estas funciones para separar en columnas los productos que vende cada comercial de la empresa. Imagina la siguiente situación:

	A	B	C	D	E	F	G	H	I	J	K
1	Resultados de búsqueda ordenados en columnas										
2											
3											
4		Cliente	Producto			Vanesa	Axel	Celso	Leonora	Rodolfo	Esmeralda
5		Vanesa	Salsa de arándanos Northwoods								
6		Axel	Buey Mishi Kobe								
7		Celso	Pez espada								
8		Leonora	Queso Cabrales								
9		Rodolfo	Queso Manchego La Pastora								
10		Esmeralda	Algas Konbu								
11		Vanesa	Cuajada de judías								
12		Axel	Salsa de soja baja en sodio								
13		Celso	Postre de merengue Pavlova								
14		Leonora	Cordero Alice Springs								
15		Rodolfo	Langostinos tigre Carnarvon								
16		Esmeralda	Pastas de té de chocolate								
17		Vanesa	Mermelada de Sir Rodney's								
18		Axel	Bollos de Sir Rodney's								
19		Celso	Pan de centeno crujiente estilo Gustaf's								
20		Leonora	Pan fino								
21		Rodolfo	Refresco Guaraná Fantástica								
22		Esmeralda	Crema de chocolate y nueces NuNuCa								
23		Vanesa	Ositos de goma Gumbär								
24		Axel	Chocolate Schoggi								
25		Celso	Col fermentada Rössle								
26		Leonora	Salchicha Thüringer								
27		Rodolfo	Arenque blanco del noroeste								

Imagina que tienes un listado de los clientes que han pasado por un comercio en los últimos días junto con los artículos que han comprado. Como cada cliente ha hecho varias compras diferentes, deseas clasificar en columnas los productos adquiridos por cada uno.

Sigue estos pasos:

1. Debajo del primer nombre, selecciona un rango de celdas vacías lo suficientemente grande para que abarque a todos los clientes y todos los productos (por ejemplo, F5:K26).
2. A continuación, pega una de las siguientes fórmulas matriciales y presiona **Ctrl + Mayús + Intro** para introducirla (de esta forma solo podrás editar la fórmula cuando tengas el mismo rango seleccionado. Esto evitará pérdidas de datos accidentales).

=SI.ERROR(INDICE(\$C\$5:\$C\$46; K.ESIMO.MENOR(SI(F\$4=\$B\$5:\$B\$46; FILA(\$C\$5:\$C\$46)-4;"")); FILA()-4));""

O

=SI.ERROR(INDICE(\$C\$5:\$C\$46;K.ESIMO.MENOR(SI(F\$4=\$B\$5:\$B\$46;FILA(\$B\$5:\$B\$46)- MIN(FILA(\$B\$5:\$B\$46))+1;"")); FILA()-4));""

Como puedes comprobar, la primera fórmula es un poco más compacta, pero la segunda es más fácil de interpretar y requiere menos modificaciones.

3. Copia la fórmula en las otras columnas. Para ello selecciona el rango de celdas donde acabas de introducir la fórmula y arrastra el controlador de relleno hacia la derecha. El resultado será algo similar a esto:

	A	B	C	D	E	F	G	H	I	J	K	L
1	Resultados de búsqueda ordenados en columnas											
2												
3												
4		Cliente	Producto			Vanesa	Axel	Celso	Leonora	Rodolfo	Esmeralda	
5		Vanesa	Salsa de arándanos Northwoods			Salsa de arándanos	Buey Mishi	Ko Pez espada	Queso Cabral	Queso Manch	Algas Konbu	
6		Axel	Buey Mishi Kobe			Cuajada de jui	Salsa de soja	Postre de mer	Cordero Alice	Langostinos ti	Pastas de té de chocola	
7		Celso	Pez espada			Mermelada di	Bollos de Sir F	Pan de center	Pan fino	Refresco Guai	Crema de chocolate y r	
8		Leonora	Queso Cabrales			Ositos de gori	Chocolate Scl	Col fermenta	Salchicha Thü	Arenque blanc	Queso gorgonzola	Telin
9		Rodolfo	Queso Manchego La Pastora			Queso Masca	Queso de cab	Cerveza Sasq	Cerveza negra	Escabeche de	Salmón ahumado	Grav
10		Esmeralda	Algas Konbu			Vino Côte de	Licor verde Cf	Carne de canç	Crema de alm	Tallarines de	Café de Malasia	
11		Vanesa	Cuajada de judías			Azúcar negra	Arenque ahun	Arenque salac	Galletas Zaan	Chocolate hol	Regaliz	
12		Axel	Salsa de soja baja en sodio									
13		Celso	Postre de merengue Pavlova									
14		Leonora	Cordero Alice Springs									
15		Rodolfo	Langostinos tigre Carnarvon									
16		Esmeralda	Pastas de té de chocolate									
17		Vanesa	Mermelada de Sir Rodney's									
18		Axel	Bollos de Sir Rodney's									

Este es un ejemplo de búsqueda avanzada que implica un conocimiento intermedio de algunas funciones de Excel. Estudiando la fórmula de dentro hacia afuera, así funciona:

La función SI

En el centro de la fórmula se encuentra SI, utilizada para obtener las posiciones de todas las repeticiones del valor de búsqueda en el rango de búsqueda.

SI compara el valor de búsqueda (F4) con cada valor del rango de búsqueda (\$B\$5:\$B\$46), y si se encuentra la coincidencia, devuelve la posición relativa de la fila. En caso contrario devuelve una cadena vacía ("").

Las posiciones relativas de las filas se calculan restando 4 de la fila ($\$C\$5:\$C\46) para que la primera fila coincida con la posición 1. Como el rango devuelto comienza en la fila cinco, se resta 4 y así sucesivamente. El resultado de esta operación es la matriz {1;2;3;4;5;6;7;8;9;10;11;12;13;14;15;16;17;18;19;20;21;22;23;24;25;26;27;28;29;30;31;32;33;34;35;36;37;38;39;40;41;42}, que va al argumento **si_verdadero** de la función SI.

El resultado de la función SI sería el siguiente:

```
{1;"";"";"";"";"";7;"";"";"";"";"";13;"";"";"";"";"";19;"";"";"";"";"";25;"";"";"";"";"";31;"";"";"";"";"";37;"";"";"";"";"";}
```

Como ves, se trata de una matriz de números (cuando hay coincidencia) y celdas vacías (cuando no se encuentra coincidencia). Si compruebas la tabla con los datos, verás que Vanesa (el valor de búsqueda en F4 aparece en las posiciones 1, 7, 13, 19, 25, 31 y 37) del rango de búsqueda en la columna B.

La función **K.ESIMO.MENOR**
 =K.ESIMO.MENOR(rango;k)

A continuación la función **K.ESIMO.MENOR** entra en acción para determinar cuál de las coincidencias debe devolverse en una celda específica.

Con la matriz ya creada vamos a trabajar en el argumento k, es decir, el valor k-ésimo más pequeño que se debe devolver. Para ello, se hace una especie de contador incremental con **FILA()-n** donde “n” es el número de la fila de la primera celda de la fórmula menos 1. En este ejemplo, se introdujo la fórmula en las celdas F5:K17, así que **FILA()-4** devuelve “1” para la celda D5 (fila 5 menos 4), “2” para la celda D6 (fila 6 menos 4), etc.

Como resultado, la función **K.ESIMO.MENOR**, extrae el primer elemento más pequeño del conjunto en la celda D5, el segundo elemento más pequeño en la celda D6 y así sucesivamente. Esto transforma la fórmula inicial tan larga y compleja en algo más sencillo como esto:

```
=SI.ERROR(INDICE($C$5:$C$46;{1});"")
```

La función **INDICE**

La función **INDICE** se utiliza para devolver el valor de un elemento de una matriz en función de su número de fila.

La función **SI.ERROR**

Finalmente se encierra la fórmula en la función **SI.ERROR** para gestionar los posibles errores, que son inevitables, pues no se puede saber cuántas repeticiones o coincidencias se devolverán para los diferentes valores de búsqueda.

Atención: Debes tener en cuenta que las referencias a las celdas son todas absolutas excepto la referencia relativa de la columna en el valor de búsqueda (F\$4), que debe cambiar en función de la posición relativa de una o varias columnas en las que se copia la fórmula para devolver coincidencias para otros valores de búsqueda.

Buscar valores con varias condiciones

¿Sabías que es posible realizar búsquedas con la función SUMAR.SI.CONJUNTO? ¿y con SUMAPRODUCTO, INDICE y COINCIDIR, BUSCARV y SUMA?

Imagina la siguiente situación:

columna aux.	Mes	Comercial	País	Producto	Nº Clientes	Ventas	Comisión
41275JuanEspañaCubo	enero-13	Juan	España	Cubo	8	1.592	563
41275JuanEspañaPrisma	enero-13	Juan	España	Prisma	8	1.088	397
41275JuanFranciaPirámide	enero-13	Juan	Francia	Pirámide	8	1.680	753
41275JuanFranciaCubo	enero-13	Juan	Francia	Cubo	9	2.133	923
41275JuanFranciaPrisma	enero-13	Juan	Francia	Prisma	10	1.610	579
41275JuanColombiaPirámide	enero-13	Juan	Colombia	Pirámide	10	1.540	570
41275JuanColombiaCubo	enero-13	Juan	Colombia	Cubo	7	1.316	428
41275JuanColombiaPrisma	enero-13	Juan	Colombia	Prisma	7	1.799	709
41275PedroEspañaPirámide	enero-13	Pedro	España	Pirámide	8	1.624	621
41275PedroEspañaCubo	enero-13	Pedro	España	Cubo	6	726	236
41275PedroEspañaPrisma	enero-13	Pedro	España	Prisma	9	2.277	966
41275PedroFranciaPirámide	enero-13	Pedro	Francia	Pirámide	6	714	221
41275PedroFranciaCubo	enero-13	Pedro	Francia	Cubo	9	2.682	1.023
41275PedroFranciaPrisma	enero-13	Pedro	Francia	Prisma	6	1.500	634
41275PedroColombiaPirámide	enero-13	Pedro	Colombia	Pirámide	7	917	403
41275PedroColombiaCubo	enero-13	Pedro	Colombia	Cubo	7	1.939	760
41275PedroColombiaPrisma	enero-13	Pedro	Colombia	Prisma	6	984	314
41275MaríaEspañaPirámide	enero-13	María	España	Pirámide	9	981	372
41275MaríaEspañaCubo	enero-13	María	España	Cubo	10	1.520	476
41275MaríaEspañaPrisma	enero-13	María	España	Prisma	6	966	330

Ahora responde a la pregunta: ¿Cuánto importe vendió Juan del producto Cubo en Colombia en febrero de 2013?

Aunque es posible hallar la respuesta a la pregunta anterior con BUSCARX o BUSCARV, te voy a mostrar otras formas de hacerlo para que utilices la que más convenga en cada caso. Utilizaré las funciones SUMAR.SI.CONJUNTO, SUMAPRODUCTO, INDICE y COINCIDIR, BUSCARV y SUMA.

Con SUMAR.SI.CONJUNTO

Esta función sirve para ¿realizar sumas con una o más condiciones, es decir, puedes sumar solo aquellas celdas que cumplen con uno o varios criterios.

La sintaxis de la función es la siguiente:

=SUMAR.SI.CONJUNTO(rango_suma; rango_criterios1; criterios1, [rango_criterios2; criterios2]; ...)

El **rango_suma** es el rango donde se encuentran los datos que se van a devolver. Luego deberías ir seleccionando cada rango y cada criterio por los que desees filtrar. Es posible incorporar a la función hasta 127 criterios diferentes, aunque no creo que llegues a necesitar tantos.

La fórmula completa sería la siguiente:

=SUMAR.SI.CONJUNTO(Ventas2[Ventas];Ventas2[Comercial];\$L\$5;Ventas2[Mes];\$L\$6;Ventas2[País];\$N\$5;Ventas2[Producto];\$N\$6)

En este caso la función SUMAR.SI.CONJUNTO devuelve solo un valor, ya que en la búsqueda que coincide con los criterios solo encuentra un dato.

Aunque esta técnica parece infalible tiene algunas limitaciones a tener en cuenta:

1. Solamente funciona cuando se están buscando valores numéricos. Si quisieras saber qué comercial hizo una venta mayor de 1.000 en Colombia, la fórmula daría error porque es imposible sumar valores de texto.
2. Si realmente estás utilizando esta función para buscar un valor, solo funcionará si filtrando todos los criterios, solamente se encuentra una coincidencia. Si se encontraran dos o más valores que cumplen todos los criterios, el resultado sería la suma de éstos.
3. No funciona en Excel 2003 y versiones anteriores. Aunque parezca que ya no existe nadie que utilice una versión tan antigua, te sorprendería saber la cantidad de consultas que me llegan de usuarios que todavía la usan.

Con SUMAPRODUCTO

Para hacer búsquedas con varios criterios también puedes utilizar la función SUMAPRODUCTO, aunque a mí me gusta más utilizar la primera (los argumentos de la función parece que quedan un poco más “limpios” en la pantalla).

Observa la fórmula:

=SUMAPRODUCTO(Ventas2[Ventas];(Ventas2[Comercial]=\$L\$5)*(Ventas2[Mes]=\$L\$6)*(Ventas2[País]=\$N\$5)*(Ventas2[Producto]=\$N\$6))

Como puedes comprobar, esta fórmula es muy parecida a la anterior. En esta estoy utilizando un criterio para cada columna.

Aunque SUMAPRODUCTO es una función muy potente y tiene muchos usos diferentes, es complicada de entender y te puede llevar un tiempo ponerla en práctica con soltura.

Al igual que en el ejemplo con SUMAR.SI.CONJUNTO, en caso de que haya varios registros que cumplan todos los criterios, el resultado que devolverá la fórmula será la suma de todos ellos.

Te aconsejo que utilices esta técnica cuando no tengas otra opción. Para buscar datos con varias condiciones es mejor usar SUMAR.SI.CONJUNTO, INDICE + COINCIDIR o SUMA.

Con SUMA

Sí, una función tan simple como SUMA también puede servir para realizar búsquedas más o menos complejas.

Observa de nuevo la fórmula:

=SUMA(Ventas2[Ventas]*(Ventas2[Comercial]=\$L\$5)*(Ventas2[Mes]=\$L\$6)*(Ventas2[País]=\$N\$5)*(Ventas2[Producto]=\$N\$6))

Como puedes comprobar es muy parecida a la anterior. De hecho, en lo único que se diferencia a simple vista es que en esta he utilizado SUMA.

Como los datos con los que trabajan los argumentos son los mismos, ambas funciones actúan de la misma forma, aunque con una pequeña diferencia: para introducir en la celda la fórmula con la función SUMA es necesario hacerlo pulsando las teclas **Ctrl + Mayusc + Intro** (fórmula matricial).

Como comprobarás, aparecen unas llaves de apertura y cierre al principio y al final de la fórmula.

Este tipo de fórmulas son las que se consideran “matriciales” o arrays, y por su estructura y complejidad, quedan fuera de este e-book. Si tienes curiosidad por ver cómo funcionan este tipo de fórmulas puedes echar un vistazo a la [Ayuda de Office](#).

Con INDICE + COINCIDIR

Como te comenté antes, aunque la función SUMAR.SI.CONJUNTO funciona perfectamente para hacer este tipo de búsquedas, falla cuando se incluyen datos de texto en el argumento de suma.

La solución a esto es utilizar INDICE + COINCIDIR en forma matricial.

Te muestro la fórmula:

=INDICE(Ventas2[Ventas];COINCIDIR(\$L\$5&\$L\$6&\$N\$5&\$N\$6;Ventas2[Comercial]&Ventas2[Mes]&Ventas2[País]&Ventas2[Producto];0))

He utilizado los mismos criterios que en las fórmulas anteriores, es decir, Comercial = Juan, Mes = febrero-13, País = Colombia y Producto = Cubo.

El mérito de la fórmula se lo lleva la función COINCIDIR. Observa cómo utilizo los nombres valorComercial, valorMes, valorPaís y valorProducto como valor a buscar. El carácter ampersand (&) es un mero concatenador de los nombres, es decir que, en realidad el valor que se está buscando es “Juan41306ColombiaCubo”.

¿Te estás preguntando qué significa 41306?. Las fechas en Excel comienzan a contar el 1 de enero de 1900. Es decir, que este día tiene el valor 1. El 2 de enero de 1900 tiene el número 2 y así sucesivamente. El valor 41306 corresponde al 1 de febrero de 2013.

Ahora que ya sabes cómo funcionan las fechas en Excel, sigo con la segunda parte de la función COINCIDIR:

Ventas2[Comercial]&Ventas2[Mes]&Ventas2[País]&Ventas2[Producto];0))

Esta parte es similar a la anterior: contiene las cuatro columnas donde se encuentran los datos. Observa que aparecen concatenadas.

CONCATENAR buscará el valor “Juan41306ColombiaCubo” en el rango de las cuatro columnas de búsqueda y devolverá la posición en la que lo encuentra.

Una vez que se le indica a INDICE en qué posición se encuentra el valor, devolverá el valor de la columna “Ventas” donde se encontró el valor.

Como la función COINCIDIR utiliza columnas no contiguas, necesitarás introducir toda la fórmula de forma matricial, es decir, pulsando **Ctrl + Mayús + Intro**.

Con BUSCARV

Todas las técnicas de este ejemplo para buscar valores que cumplen varios criterios, funcionan correctamente. Esta última, utilizando BUSCARV, también funciona perfectamente, aunque se sirve de una columna de datos auxiliar.

Utilizar una columna de datos auxiliar, crea una columna en la que cada registro es único, ya que se sirve de referencia de otras columnas de la tabla.

Te muestro cómo trabajar con columnas auxiliares:

En primer lugar, debes crear una columna en blanco preferiblemente en la parte izquierda de la tabla. En esta columna inserta la fórmula que concatena los cuatro criterios por los que se va a hacer la búsqueda:

=C7&D7&E7&F7

Búsqueda con varios criterios			
	columna aux.	Mes	Comercial
7	41275JuanEspañaCubo	enero-13	Juan
8	41275JuanEspañaPrisma	enero-13	Juan
9	41275JuanFranciaPirámide	enero-13	Juan
10	41275JuanFranciaCubo	enero-13	Juan
11	41275JuanFranciaPrisma	enero-13	Juan

Para simplificar las cosas, selecciona los datos de toda la tabla resultante y asígnale el nombre de "Datos2". Esto no es necesario, pero mejora la legibilidad y permite agregar nuevos datos a la tabla cuando se requieran.

Te muestro la fórmula a introducir:

=BUSCARV(L6&L5&N5&N6;B7:I329;7;0)

A estas alturas de lectura del libro, seguro que no hace falta que te explique mucho de la fórmula anterior, pero por si acaso, lo haré:

El valor buscado se compone de la concatenación de las cuatro columnas donde se encuentran los criterios. Este valor buscado se busca en la tabla que comienza con la columna auxiliar. Como se trata de un valor único, no hay lugar a error. El resultado que devuelve es el dato de la séptima columna a partir del principio de la tabla.

Sencillo, ¿Verdad?

Qué técnica utilizar en cada caso

Aunque ya conoces varios métodos para buscar valores con varios criterios, seguro que te preguntas cuándo es mejor utilizar uno que otro:

- SUMAR.SI.CONJUNTO cuando necesites extraer un valor numérico.
- INDICE + COINCIDIR para buscar textos o valores mixtos.
- BUSCARV si tienes espacio para añadir una columna auxiliar en la parte izquierda de la tabla.
- SUMAPRODUCTO puedes utilizarla como último recurso, porque casi siempre lograrás tu objetivo con las tres anteriores.

Recuerda que hay muchas maneras de hacer la misma cosa en Excel. No existe una forma correcta o incorrecta. Solamente utiliza la que te funcione.

Como consejo te diré que no pierdas el tiempo buscando la fórmula perfecta para solucionar cada caso. Tu jefe seguro que sólo te exigirá resultados sin importarle cómo has llegado a ellos.

Obtener el valor máximo de una búsqueda

Ya sabes que utilizar BUSCARX o BUSCARV son las mejores formas de extraer información de una tabla según uno o varios criterios. Pero todos los datos extraídos hasta ahora se encontraban dentro de una sola columna. ¿Qué ocurre si necesitas evaluar información en más de una columna? Imagina que tienes una tabla con varios comerciales en la columna de la izquierda y las ventas mensuales de cada uno, mes a mes en las columnas de la derecha. Estás buscando las ventas de todo el año.

Según lo visto hasta ahora, podrías sumar todos los resultados de un montón de funciones BUSCARV, pero al final quedaría una fórmula larguísima que sería muy difícil de entender.

Comercial	Ene.	Feb.	Mar.	Abr.	May.	Jun.	Jul.	Ago.	Sep.	Oct.	Nov.	Dic.
Manuel	22	22	24	25	26	28	30	32	33	33	36	36
Alfonso	10	10	11	11	12	12	13	13	14	15	15	15
Eduardo	26	26	27	27	27	27	29	30	30	32	32	34
María	18	19	20	20	20	21	22	23	23	24	25	25
Adriana	19	20	22	24	25	25	25	27	29	30	30	31
Javier	16	17	18	18	19	20	22	24	25	26	27	28
Adolfo	15	16	17	17	18	18	18	19	19	20	20	22
Sergio	10	11	11	11	12	13	13	14	14	14	15	15
Juan Antonio	21	21	21	23	24	24	25	26	27	29	30	32
Pedro	25	26	27	28	29	29	29	30	31	34	37	40
Guillermo	22	24	25	26	28	28	29	31	34	35	35	35
Laura	23	24	24	24	25	26	27	28	30	33	35	37
Silvia	19	20	20	21	22	24	24	26	27	27	28	30

Para simplificar la extracción de datos de la tabla, puedes utilizar una lista de campos en vez del número de columna a devolver. Por ejemplo, la siguiente fórmula:

=SUMA (BUSCARV (Valor; Tabla de datos; {2\3\4}; FALSO)

Devolverá la suma de los valores de las columnas 2, 3 y 4. Recuerda que para introducir correctamente la fórmula debes pulsar **Ctrl + Mayúsc + Intro**,

Cuando introduzcas matrices presta atención a los separadores de valores, pues según la configuración regional de tu país, puede ser diferente a la mostrada aquí.

Si nos pidieran averiguar cuál fue la venta más alta realizada por Javier durante el año la respuesta sería 28.

Mira la fórmula:

=MAX(BUSCARV(Q6;\$B\$6:\$N\$18;{2\3\4\5\6\7\8\9\10\11\12\13};FALSO))

La fórmula obtiene los datos de todos los meses de Javier (columnas de la 2 a la 13). Estos datos son traspasados a la función MAX, que es la encargada de devolver el valor máximo.

Utilizando una fórmula similar también podemos hallar, por ejemplo, el promedio de ventas que tuvo Javier durante el año. Tendríamos que utilizar la fórmula:

=PROMEDIO(BUSCARV(Q6;\$B\$6:\$N\$18;{2\3\4\5\6\7\8\9\10\11\12\13};FALSO))

Como puedes apreciar funciona exactamente igual que la función anterior. La única diferencia es que los datos obtenidos de todos los meses se trasladan a la función PROMEDIO.

Varias técnicas de búsqueda de datos en tablas de doble entrada

Hasta ahora te he mostrado las mejores formas de hacer búsquedas en una y en varias columnas para obtener uno o varios datos ubicados en otra columna de la misma fila.

En esta sección aprenderás a hacer búsquedas en tablas de doble entrada, es decir que buscarás un dato que coincida con dos criterios ubicados en una fila y una columna. Observa la siguiente imagen para hacerte una mejor idea:

Comercial	Ene.	Feb.	Mar.	Abr.	May.	Jun.	Jul.	Ago.	Sep.	Oct.	Nov.	Dic.
Manuel	22	22	24	25	26	28	30	32	33	33	36	36
Alfonso	10	10	11	11	12	12	13	13	14	15	15	15
Eduardo	26	26	27	27	27	27	29	30	30	32	32	34
María	18	19	20	20	20	21	22	23	23	24	25	25
Adriana	19	20	22	24	25	25	25	27	29	30	30	31
Javier	16	17	18	18	19	20	22	24	25	26	27	28
Adolfo	15	16	17	17	18	18	18	19	19	20	20	22
Sergio	10	11	11	11	12	13	13	14	14	14	15	15
Juan Antonio	21	21	21	23	24	24	25	26	27	29	30	32
Pedro	25	26	27	28	28	29	29	30	31	34	37	40
Guillermo	22	24	25	26	28	28	29	31	34	35	35	35
Laura	23	24	24	24	25	26	27	28	30	33	35	37
Silvia	19	20	20	21	22	24	24	26	27	27	28	30

Encontrar el valor que coincide con la fila y la columna

Aunque parezca complicado hacer una búsqueda de este tipo, no es así. A continuación, te muestro varias formas de hacerlo.

Antes de nada debes tener en cuenta

Para trabajar más relajadamente he creado algunos nombres de rango:

- valComercial y valMes, se refieren al comercial y el mes que se va a buscar.
- IstComercial y IstMes se refieren a los nombres de los comerciales y de los meses.
- tblEntrada contiene únicamente los datos de ventas de todos los meses.

	Comercial	Ene.	Feb.	Mar.	Abr.	May.	Jun.	Jul.	Ago.	Sep.	Oct.	Nov.	Dic.
IstComercial	Manuel	22	22	24	25	26	28	30	32	33	33	36	36
	Alfonso	10	10	11	11	12	12	13	13	14	15	15	15
	Eduardo	26	26	27	27	27	27	29	30	30	32	32	34
	María	18	19	20	20	20	21	22	23	23	24	25	25
	Adriana	19	20	22	24	25	25	25	27	29	30	30	31
	Javier	16	17	18	18	19	20	22	24	25	26	27	28
	Adolfo	15	16	17	17	18	18	18	19	19	20	20	22
	Sergio	10	11	11	11	12	13	13	14	14	14	15	15
	Juan Antonio	21	21	21	23	24	24	25	26	27	29	30	32
	Pedro	25	26	27	28	29	29	29	30	31	34	37	40
	Guillermo	22	24	25	26	28	28	29	31	34	35	35	35
	Laura	23	24	24	24	25	26	27	28	30	33	35	37
	Silvia	19	20	20	21	22	24	24	26	27	27	28	30

Con INDICE y COINCIDIR

Si sabes cuál es el número de fila y de columna donde se encuentran los valores a buscar, puedes utilizar la función INDICE para buscar la posición del elemento en la intersección y luego usar la función COINCIDIR para devolver el valor que se encuentra en esa posición.

Combinando las dos funciones:

=INDICE(tblEntrada;COINCIDIR(valComercial;IstComercial;0);COINCIDIR(valMes;IstMes;0))

Esta será la fórmula que deberás utilizar para encontrar la venta que hizo un determinado comercial en un mes en concreto.

Para interpretar correctamente esta y otras fórmulas debes hacerlo desde dentro hacia afuera:

- COINCIDIR(valComercial;IstComercial;0) va a buscar la posición del vendedor definido en valComercial dentro de la lista IstComercial. Por ejemplo, si estás buscando a “Eduardo”, la función devolverá 3.
- COINCIDIR(valMes;IstMes;0) buscará el valor del mes que se encuentra definido en valMes dentro de la lista IstMes.
- =INDICE(tblEntrada;COINCIDIR(...);COINCIDIR(...)). Esta parte de la fórmula se encarga de buscar dentro de la tabla tblEntrada, los valores devueltos por las funciones COINCIDIR.

Con SUMAPRODUCTO

Este es el método que más me gusta.

Puedes utilizar la función SUMAPRODUCTO para obtener el valor de la intersección con la siguiente fórmula:

=SUMAPRODUCTO((IstComercial=valComercial)*(IstMes=valMes);tblEntrada)

El funcionamiento es muy simple: cuando escribes

(IstComercial=valComercial)*(IstMes=valMes)

SUMAPRODUCTO genera un montón de ceros en cada celda de la tabla excepto en la intersección de los dos criterios que le proporcionas, donde se genera un 1.

Observa la siguiente imagen para que entiendas cómo funciona SUMAPRODUCTO:

	FALSO	FALSO	VERDADERO	FALSO								
FALSO	0	0	0	0	0	0	0	0	0	0	0	0
FALSO	0	0	0	0	0	0	0	0	0	0	0	0
FALSO	0	0	0	0	0	0	0	0	0	0	0	0
FALSO	0	0	0	0	0	0	0	0	0	0	0	0
VERDADERO	0	0	1	0	0	0	0	0	0	0	0	0
FALSO	0	0	0	0	0	0	0	0	0	0	0	0
FALSO	0	0	0	0	0	0	0	0	0	0	0	0
FALSO	0	0	0	0	0	0	0	0	0	0	0	0
FALSO	0	0	0	0	0	0	0	0	0	0	0	0
FALSO	0	0	0	0	0	0	0	0	0	0	0	0
FALSO	0	0	0	0	0	0	0	0	0	0	0	0
FALSO	0	0	0	0	0	0	0	0	0	0	0	0
FALSO	0	0	0	0	0	0	0	0	0	0	0	0

Como solamente existe un valor que coincide con cada uno de los dos criterios, la función SUMAPRODUCTO devolverá el valor que coincida con ambos.

SUMAPRODUCTO es una función avanzada que no solo puedes utilizar para sumar el producto de varias columnas. Con un poco de imaginación puedes emplearla en otros cálculos que no tienen nada que ver con el aspecto puramente aritmético.

Con BUSCAR y COINCIDIR

Otra forma igual de original que la anterior es con la combinación de BUSCARV y COINCIDIR.

=BUSCARV(valComercial;\$B\$7:\$N\$19;COINCIDIR(valMes;lstMes;0)+1;FALSO)

Este método busca el comercial en la primera columna de la tabla y devuelve el valor de la columna que coincida con el criterio de mes que le proporcionas.

Como ves, al resultado de la función COINCIDIR debes sumarle 1 ya que la tabla en la que se busca, contiene una columna más en la parte izquierda (la de los comerciales), que la función BUSCARV tiene también en cuenta

¿Cuál de los tres métodos es mejor?

La verdad es que las tres formas de hacer búsquedas en tablas de doble entrada son efectivas. Todas funcionan en cualquier situación, aunque si tuviese que elegir una, me quedaría con la segunda. La única razón es que es más fácil de entender por otros usuarios que puedan necesitar modificar la fórmula.

Cómo encontrar la 2ª, 3ª o nª coincidencia

Hasta ahora, todo lo que te he mostrado a lo largo de este e-book han sido algunas funciones para encontrar siempre la primera coincidencia con el criterio buscado. Pero, ¿qué ocurre si necesitas buscar la segunda o tercera coincidencia?

Observa la imagen de debajo. Si lo que necesitas es encontrar la cantidad de ventas netas que hizo Pedro por segunda vez, la función BUSCARV, utilizada como hasta ahora, no sirve de mucho.

Comercial	Nº clientes	Ventas netas	Comisiones
Sergio	8	1.592,00	562,77
María	8	1.088,00	396,90
Sergio	8	1.680,00	752,64
María	9	2.133,00	922,74
Adriana	10	1.610,00	579,12
Pedro	10	1.540,00	569,80
Pedro	7	1.316,00	427,57
Sergio	7	1.799,00	708,81
María	8	1.624,00	621,34
Pedro	6	726,00	235,88
Sergio	9	2.277,00	965,68
María	6	714,00	220,98
Pedro	9	2.682,00	1.023,18

Para resolver este problema te desarrollo dos soluciones totalmente diferentes (aunque se puede hacer de varias formas más):

Con columna de ayuda

La forma más sencilla de resolver esta situación es generar valores únicos para la columna "Comercial". Para ello debes crear una columna de ayuda (donde mejor te venga, pero te sugiero que sea en la parte izquierda) que contendrá valores únicos consistentes en el nombre del comercial y el número de repetición.

#Comercial	Comercial	Nº clientes	Ventas netas	Comisiones
Sergio1	Sergio	8	1.592,00	562,77
María1	María	8	1.088,00	396,90
Sergio2	Sergio	8	1.680,00	752,64
María2	María	9	2.133,00	922,74
Adriana1	Adriana	10	1.610,00	579,12
Pedro1	Pedro	10	1.540,00	569,80
Pedro2	Pedro	7	1.316,00	427,57
Sergio3	Sergio	7	1.799,00	708,81
María3	María	8	1.624,00	621,34
Pedro3	Pedro	6	726,00	235,88
Sergio4	Sergio	9	2.277,00	965,68
María4	María	6	714,00	220,98
Pedro4	Pedro	9	2.682,00	1.023,18

Aunque existen muchas formas de generar valores únicos a partir de datos repetidos no me voy a extender mucho en ello.

La forma más sencilla es utilizando la función CONTAR.SI de la siguiente forma:

=C6&CONTAR.SI(\$C\$6:C6;C6)

Observa cómo la referencia (\$C\$6:C6;C6) es una referencia mixta (mezcla entre relativa y absoluta). Gracias a esto es posible obtener los valores correlativos Sergio1, Sergio2...

Una vez que has conseguido crear la lista de valores únicos, el resto es muy fácil, porque lo has estado viendo anteriormente. Puedes aplicar la siguiente fórmula:

=BUSCARV(\$H7&COLUMNAS(\$I\$5:I\$5);TablaDatos3;4;FALSO)

Sin columna de ayuda

En algunas ocasiones la disposición de los datos y el espacio en la hoja de cálculo juega un papel muy importante. Es posible que no tengas espacio para crear una columna auxiliar o sencillamente no quieras hacerlo.

En este caso la fórmula a utilizar para encontrar la 2ª, 3ª o nª coincidencia es algo más compleja:

=SI.ERROR(INDICE(\$D\$6:\$D\$18;K.ESIMO.MENOR(SI(\$B\$6:\$B\$18=\$H7;FILA(\$B\$6:\$B\$18)-FILA(\$B\$5));I\$6));"")

Atención: se trata de una fórmula matricial y por tanto debes introducirla presionando **Ctrl + Mayús + Intro**.

Buscar la última coincidencia de un valor

Algo bastante común, además de buscar la n repetición de un dato, es buscar la última vez que se repite.

Con BUSCARX es relativamente sencillo. Solo hay que invertir el orden de búsqueda (lo vimos anteriormente), pero existen formas alternativas.

Ya has visto que para encontrar la segunda repetición tenemos que buscar "Sergio2" en la columna de ayuda. Para encontrar la última repetición, habría que buscar "Sergio x ", donde x es el número de veces que aparece en el listado.

La función para este caso es CONTAR.SI. Utilizando

=BUSCARV(\$H14&CONTAR.SI(\$C\$6:\$C\$18;\$H7);TablaDatos3;4;FALSO)

Estarías buscando la última repetición del comercial.

CONTAR.SI(\$C\$6:\$C\$18;\$H7) cuenta el número de veces que se repite el valor en la columna C y ese resultado se concatena con nombre del comercial creando el valor único de búsqueda.

Algunos trucos mostrados en el ebook

Hasta aquí te he mostrado cómo utilizar BUSCARX, BUSCARV y algunas otras funciones. En esta última parte te hablaré de los trucos que he venido utilizando en el ebook.

Utiliza rangos con nombres

¿Qué sensación te da cuando ves una fórmula llena de símbolos de dólar y de referencias de celdas? Cuando la fórmula contiene una o dos funciones no es difícil de interpretar pero cuando se comienzan a anidar más de tres o cuatro.... La mejor opción para hacer que a simple vista sepas cómo funciona una fórmula

es [utilizar nombres](#). Por ejemplo, BUSCARV(Tasa;Tabla_Comisiones;4;FALSO) es mucho más legible que BUSCARV(G5,\$B\$5:\$G\$17,3,FALSO). ¿No crees?

Utiliza referencias de celdas absolutas siempre que sea posible

Cuando se deben crear varias fórmulas similares, la gran mayoría de usuarios escriben la primera y luego la arrastran hacia abajo. Si no se utilizan los tipos de referencias adecuadas puede que no se encuentre en la tabla el dato que buscas, ya que las referencias de cada celda cambian según vas arrastrando las fórmulas

Siempre que puedas utiliza referencias absolutas en el argumento **Tabla_de_datos** (por ejemplo \$A\$2:\$C\$500) para que las celdas permanezcan fijas aunque arrastres la fórmula.

Utiliza tablas

Utilizar rangos con nombres y referencias absolutas es la opción ideal cuando la tabla de datos a la que se refieren las funciones contiene siempre la misma cantidad de datos. Si deseas utilizar un conjunto de datos que se irán haciendo más grandes con el paso del tiempo te aconsejo convertir este rango en una tabla. Las tablas ajustan sus referencias automáticamente al rango de datos aunque aumente o disminuya.

Gestiona los errores

A no ser que utilices BUSCARX para realizar tus búsquedas, con ninguna de las demás funciones por sí solas, podrás gestionar los errores que puedan producirse.

Si con BUSCARV quieres encontrar un valor que no existe, se producirá un error. Gestionar estos errores es fundamental, sobre todo si tienes pensado distribuir tu trabajo a usuarios menos expertos. Para gestionar estos errores puedes utilizar la función SI.ERROR (para Excel 2007) de esta forma:

SI.ERROR(BUSCARV(...);"Valor no encontrado")

Si todavía no has actualizado a las últimas versiones de Excel te propongo que utilices la siguiente estructura:

=SI (SI.ERROR (BUSCARV (...));"Valor no encontrado";BUSCARV(...))

La fórmula anterior es fácil de entender, pero no es la mejor opción ya que Excel necesita calcular BUSCARV dos veces. Para grandes cantidades de datos te recomiendo que utilices la función CONTAR.SI:

SI(CONTAR.SI(columna; valor a BUSCAR)>0;BUSCARV(...);"Valor no encontrado").

Utiliza BUSCARV solo cuando lo necesites

Sí, ya sé que es algo extraño, dado que estás leyendo un libro sobre búsquedas y la función de la que más se habla es de BUSCARV. En muchos casos, según tus necesidades, funciones como SUMAR.SI o SUMAPRODUCTO pueden sustituirla de forma efectiva. Por ejemplo: la fórmula =SUMAR.SI(rango;criterio;rango_suma) puede devolver como resultado el mismo valor que =BUSCARV(valor_buscado; tabla_datos; 2;FALSO).

De forma similar, si quieres saber si un valor se encuentra en una lista de números, puedes utilizar la función CONTAR.SI.

¿Quieres aprender más sobre Excel?

Si tienes preguntas acerca del contenido o los ejemplos de este libro, puedes resolver todas tus dudas en el [foro de Ayuda Excel](#), donde mi equipo y yo estaremos encantados de echarte una mano.

Descarga de archivos

Descarga todos los ejemplos propuestos en el ebook en el siguiente enlace:

<http://ayudaexcel.com/archivos-de-trabajo-domina-las-funciones-de-busqueda/>